

**С. Е. Савотченко,
Т.Г. Кузьмичева**

**Методы решения
математических задач в
*Maple***

*Рекомендовано Редакционно-издательским советом
Белгородского государственного университета в качестве
учебного пособия для студентов социально-психологического и
естественно-географического факультетов
Белгородского государственного университета*

Белгород
2001

ББК 22.11

С 13

**Рецензенты: Чеканов Н.А. д.ф.-м.н., профессор БелГУ;
Сурков Э.П. к.ф.-м.н., профессор БГСХА.**

Савотченко С.Е., Кузьмичева Т.Г.

С13 Методы решения математических задач в *Maple*: Учебное пособие – Белгород: Изд. Белаудит, 2001. – 116 с.

ISBN 5-7414-0046-9

Данная книга является учебным пособием по дисциплинам «Математика и информатика», «Информационные технологии». Пособие представляет собой практическое руководство по изучению возможностей пакета аналитических вычислений *Maple*. Подробные теоретические сведения чередуются с практическими заданиями. Последовательное изучение тем и выполнение заданий позволит шаг за шагом освоить основные приемы работы в математической системе *Maple*.

Учебное пособие предназначено для студентов 1 и 2 курсов социально-психологического и естественно-географического факультетов университета, а также для аспирантов и научных работников, использующих математические методы и модели в естественнонаучных исследованиях.

ББК 22.11

ISBN 5-7414-0046-9

© Савотченко С.Е., Кузьмичева Т.Г.,
2001

Предисловие

Данное пособие представляет собой практическое руководство по изучению возможностей пакета аналитических вычислений *Maple*.

Следует отметить, что эта книга ни в коей мере не может считаться описанием программного продукта *Maple*. Она предназначена в первую очередь для обучения студентов решению математических задач на персональном компьютере при помощи *Maple*. Задачи и упражнения, приведенные в качестве примеров и практических заданий, соответствуют программам по курсу общей математики для студентов естественнонаучных и социально-психологических специальностей вузов.

Структура пособия: пособие состоит из семи тем. В каждой теме содержатся:

- теоретическая часть – в ней приведено описание изучаемых команд *Maple*;
- практические задания – подробное пошаговое описание действия команд *Maple* на конкретных примерах по математике; эти задания предназначены для выполнения студентами под руководством преподавателя;
- контрольные задания – задачи и упражнения для самостоятельного выполнения студентами;
- контрольные вопросы – предназначены для закрепления теоретического материала.

Практические задания можно выполнять на лабораторных работах по одной теме на каждую лабораторную работу.

Выполнение заданий по темам №1 и 2 рассчитано на 2-х часовые занятия, выполнение заданий по остальным темам предполагает 4-х часовые аудиторские занятия, включая выполнение контрольных заданий.

Материал, содержащийся в темах №1-5, соответствует программам по математике для студентов 1 курса, а материал, содержащийся в темах №6-7, рассчитан для студентов 2 курса специальностей с более углубленными программами по математике, например, «химия».

Авторы выражают благодарность Воронову В.П., благодаря которому была начата работа над данной книгой.

I. Структура окна *Maple*. Арифметические операции, числа, константы и стандартные функции. Элементарные преобразования математических выражений

1. Структура окна *Maple*.
2. Арифметические операции, целые и рациональные числа и константы *Maple*.
3. Синтаксис команд. Стандартные функции.
4. Преобразования математических выражений.

§1. Структура окна *Maple*

Maple – это пакет для аналитических вычислений на компьютере, содержащий более двух тысяч команд, которые позволяют решать задачи алгебры, геометрии, математического анализа, дифференциальных уравнений, статистики, математической физики.

Для того, чтобы запустить *Maple*, необходимо в *Главном меню Windows* выбрать в группе *Программы* название данного приложения: *Maple*.

Maple представляет собой типичное окно *Windows*, которое состоит из *Строки названия*, *Основного меню*, *Панели инструментов*, *Рабочего поля* и *Строки состояния*, а также *Линейки* и *Полос прокрутки*.

Вид фрагмента окна *Maple 6*, содержащего *Строку названия*, *Основное меню*, *Панель инструментов*:



Пункты *Основного меню*:

File (Файл) – содержит стандартный набор команд для работы с файлами, например: сохранить файл, открыть файл, создать новый файл и т.д.

Edit (Правка) – содержит стандартный набор команд для редактирования текста, например: копирование, удаление выделенного текста в буфер обмена, отмена команды и т.д.

View (Вид) – содержит стандартный набор команд, управляющих структурой окна *Maple*.

Методы решения математических задач в *Maple*

Insert (Вставка) – служит для вставки полей разных типов: математических текстовых строк, графических двух и трехмерных изображений.

Format (Формат) – содержит команды оформления документа, например: установка типа, размера и стиля шрифта.


Options (Параметры) – служит для установки различных параметров ввода и вывода информации на экран, принтер, например, таких как качество печати.


Windows (Окно) – служит для перехода из одного рабочего листа в другой.

Help (Справка) – содержит подробную справочную информацию о *Maple*.

Работа в *Maple* проходит в режиме сессии – пользователь вводит предложения (команды, выражения, процедуры), которые воспринимаются условно и обрабатываются *Maple*. Рабочее поле разделяется на три части:

- 1) область ввода - состоит из командных строк. Каждая командная строка начинается с символа >;
- 2) область вывода - содержит результаты обработки введенных команд в виде аналитических выражений, графических объектов или сообщений об ошибке;
- 3) область текстовых комментариев - содержит любую текстовую информацию, которая может пояснить выполняемые процедуры. Текстовые строки не воспринимаются *Maple* и никак не обрабатываются.

Для того, чтобы переключить командную строку в текстовую, следует на *Панели инструментов* нажать мышью на кнопку 

Обратное переключение текстовой строки в командную осуществляется нажатием на *Панели инструментов* на кнопку 

Задание 1.

1. Запустите *Maple*.
2. После запуска *Maple* первая строка оказывается командной. Переведите ее в текстовую. Наберите в этой строке: «Лабораторная работа №1» и название темы. Перейдите на следующую строку, нажав *Enter*.
3. В новой строке наберите «Выполнил студент » и свою фамилию. Нажмите *Enter*.
4. На следующей строке наберите «Задание №1».

Методы решения математических задач в *Maple*

5. Сохраните свой файл на дискете. Для этого в меню **File** выберите пункт **Save** и наберите имя вашего файла в виде: Фамилия_1, где указывается ваша фамилия и 1 – номер лабораторной работы.
6. После этого в следующей строке наберите текст: «Файл с заданиями лабораторной работы №1 сохранен под именем: Фамилия_N».

В дальнейшем выполнение каждой лабораторной работы должно оформляться таким способом. В начале каждой лабораторной работы следует набирать текст: «Лабораторная работа N», N – номер темы. Выполнение каждого задания следует начинать с текстового комментария: «Задание N». Для правильности вычислений перед выполнением каждого пункта задания следует выполнять команду **restart**. Перед выполнением контрольных заданий следует набирать в текстовом режиме «Контрольные задания». После окончания выполнения работы необходимо сохранить файл со всеми выполненными заданиями на диск. Имя вашего файла набирается в виде: Фамилия_N, где указывается ваша фамилия и N – номер темы.

§2. Арифметические операции.

Целые и рациональные числа, константы в *Maple*

Математические константы и арифметические операции.

Основные математические константы:

Pi – число π ; **I** – мнимая единица i ; **infinity** – бесконечность;
Gamma – константа Эйлера; **true, false** – логические константы, обозначающие истинность и ложность высказывания.

Знаки арифметических операций:

+ - сложение; **-** - вычитание;

***** - умножение; **/** - деление;

^ - возведение в степень; **!** – факториал.

Знаки сравнения: **<**, **>**, **>=**, **<=**, **<>**, **=**.

Комплексные, целые и рациональные числа.

Числа в *Maple* бывают действительные (real) и комплексные (complex). Комплексное число записывается в алгебраической форме $z=x+iy$, и в командной строке такая запись должна выглядеть так:

> z:=x+I*y;

Вещественные числа разделяются на целые и рациональные. Целые числа (integer) выражаются цифрами в десятичной записи. Рациональные числа могут быть представлены в 3-х видах:

Методы решения математических задач в *Maple*

- 1) рациональной дроби с использованием оператора деления, например: $28/70$;
- 2) с плавающей запятой (float), например: 2.3 ;
- 3) в показательной форме, например: $1,602*10^{(-19)}$ означает $1,602*10^{-19}$.

Для того, чтобы получить рациональное число не в точной форме, а в виде приближенного значения (числа с плавающей запятой), следует дописывать к целой части числа $.0$. Пример:

$> 75/4;$

$$\frac{75}{4}$$

$> 75/4.0;$

18.75000000

В *Maple* можно записать буквы греческого алфавита в полиграфическом виде. Для этого в командной строке набирается название греческой буквы. Например, буква α получится, если набрать **alpha**.

Таблица строчных греческих букв и их названий:

α - alpha	η - eta	μ - mu	ϕ - phi
β - beta	θ - theta	ξ - xi	χ - chi
γ - gamma	ι - ita	π - pi	ψ - psi
δ - delta	κ - kappa	ρ - rho	ω - omega
ε - epsilon	λ - lambda	σ - sigma	
ζ - zeta	ν - nu	υ - upsilon	

Заглавные греческие буквы можно записать, если набирать название греческой буквы с заглавной, например, чтобы получить Ω , следует набрать **Omega**. Греческие буквы также можно набирать с помощью специального меню.

Задание 2.

1. Перейдите в текстовый режим и наберите «Задание №2». После не забудьте перейти в режим командной строки.
2. Вычислите значение $\frac{\sqrt{6+2\sqrt{5}} - \sqrt{6-2\sqrt{5}}}{\sqrt{3}}$. Для этого в командной строке наберите:
 $> (\text{sqrt}(6+2*\text{sqrt}(5)) - \text{sqrt}(6-2*\text{sqrt}(5))) / \text{sqrt}(3);$

и нажмите *Enter*. В результате получится точное значение $\frac{2}{3}\sqrt{3}$.

3. Наберите формулы $\omega = \frac{\theta}{t}$ и $|f(x) - \delta| < \varepsilon$. Для этого в командной строке наберите:
> **omega=theta/t; abs(f(x)-delta)<epsilon;**
нажмите *Enter*.

§3. Синтаксис команд. Стандартные функции

Синтаксис команд.

Стандартная команда *Maple* состоит из имени команды и ее параметров, указанных в круглых скобках: **command(p1, p2, ...)**. В конце каждой команды должен быть знак (;) или (:). Разделитель (;) означает, что в области вывода после выполнения этой команды будет сразу виден результат. Разделитель (:) используется для отмены вывода, то есть когда команда выполняется, но ее результат на экран не выводится.

Символ процента (%) служит для вызова предыдущей команды. Этот символ играет роль краткосрочной замены предыдущей команды с целью сокращения записи. Пример использования (%):

> **a+b;**

$a+b$

> **%+c;**

$a+b+c$.

Для присвоения переменной заданного значения используется знак присвоить (:=).

Когда программа *Maple* запускается, она не имеет ни одной команды, полностью загруженной в память. Большая часть команд имеют указатели их нахождения, и при вызове они загружаются автоматически. Другие команды находятся в стандартной библиотеке и перед выполнением обязательно должны быть вызваны командой **readlib(command)**, где **command** – имя вызываемой команды. Остальная часть процедур *Maple* содержится в специальных библиотеках подпрограмм, называемых пакетами. Пакеты необходимо подгружать при каждом запуске файла с командами из этих библиотек. Имеется два способа вызова команды из пакета:

- 1) можно загрузить весь пакет командой **with(package)** где **package** – имя пакета;

Методы решения математических задач в *Maple*

- 2) вызов какой-нибудь одной команды **command** из любого пакета **package** можно осуществить, если набрать команду в специальном формате:

> **package[command] (options) ;**

где вначале записывается название пакета **package**, из которого надо вызвать команду, а затем в квадратных скобках набирается имя самой команды **command**, и после чего в круглых скобках следуют параметры **options** данной команды.

К библиотекам подпрограмм *Maple* относятся, например, следующие пакеты: **linalg** – содержит операции линейной алгебры; **geometry** – решение задач планиметрии; **geom3d** – решение задач стереометрии; **student** – содержит команды, позволяющие провести поэтапное решение задачи в аналитическом виде с промежуточными вычислениями.

Стандартные функции.

Стандартные функции <i>Maple</i>	
Математическая запись	Запись в <i>Maple</i>
e^x	exp (x)
$\ln x$	ln (x)
$\lg x$	log10 (x)
$\log_a x$	log [a] (x)
\sqrt{x}	sqrt (x)
$ x $	abs (x)
$\sin x$	sin (x)
$\cos x$	cos (x)
$\operatorname{tg} x$	tan (x)
$\operatorname{ctg} x$	cot (x)
$\sec x$	sec (x)
$\operatorname{cosec} x$	csc (x)
$\arcsin x$	arcsin (x)
$\arccos x$	arccos (x)
$\operatorname{arctg} x$	arctan (x)
$\operatorname{arcctg} x$	arccot (x)
$\operatorname{sh} x$	sinh (x)
$\operatorname{ch} x$	cosh (x)

Методы решения математических задач в Maple

thx	tanh (x)
cthx	coth (x)
$\delta(x)$ - функция Дирака	Dirac (x)
$\theta(x)$ - функция Хевиссайда	Heaviside (x)

Maple содержит огромное количество специальных функций, таких, как Бесселевы функции, Эйлеровы бета- и гамма – функции, интеграл ошибок, эллиптические интегралы, различные ортогональные полиномы.

С помощью функции **exp (x)** определяется число $e=2.718281828\dots$ посредством записи **exp (1)**.

Задание 3.

1. Перейдите в текстовый режим и наберите «Задание №3». После не забудьте перейти в режим командной строки.

2. Вычислите $\cos \frac{\pi}{3} + \operatorname{tg} \frac{14\pi}{3}$ Для этого наберите в командной строке:

```
> cot (Pi/3) +tan (14*Pi/3) ;
```

Нажмите *Enter*. В результате в области вывода должно появиться число: $-\frac{2}{3}\sqrt{3}$.

3. Вычислите $\sin^4 \frac{\pi}{8} + \cos^4 \frac{3\pi}{8} + \sin^4 \frac{5\pi}{8} + \cos^4 \frac{7\pi}{8}$. Для этого наберите в командной строке:

```
> combine ( (sin (Pi/8) ) ^4+ (cos (3*Pi/8) ) ^4+  
(sin (5*Pi/8) ) ^4+ (cos (7*Pi/8) ) ^4) ;
```

Нажмите *Enter*. (значение команды **combine** – преобразовывать выражения, например, со степенями). В результате в области вывода должно появиться число: $\frac{3}{2}$.

§4. Преобразование математических выражений

Maple обладает широкими возможностями для проведения аналитических преобразований математических формул. К ним относятся такие операции, как приведение подобных, разложение на

множители, раскрытие скобок, приведение рациональной дроби к нормальному виду и многие другие.

Выделение частей выражений.

Математическая формула, над которой будут производиться преобразования, записывается в следующей форме: **> eq:=exp1=exp2;** где **eq** – произвольное имя выражения, **exp1** – условное обозначение левой части формулы, **exp2** – условное обозначение правой части формулы.

Выделение правой части выражения осуществляется командой **rhs (eq)**, выделение левой части выражения – командой **lhs (eq)**. Рассмотрим пример:

> eq:=a^2-b^2=c;

$$eq := a^2 - b^2 = c$$
> lhs (eq) ;

$$a^2 - b^2$$
> rhs (eq) ;

Если задана рациональная дробь вида a/b , то можно выделить ее числитель и знаменатель с помощью команд **numer** и **denom**, соответственно. Пример:

> f:=(a^2+b)/(2*a-b) ;

$$f := \frac{a^2 + b}{2a - b}$$
> numer (f) ;

$$a^2 + b$$
> denom (f) ;

$$2a - b$$

Тождественные преобразования выражений.

Раскрытие скобок выражения **eq** осуществляется командой **expand (eq)**. Пример:

> eq:=(x+1)*(x-1)*(x^2-x+1)*(x^2+x+1) ;

$$eq := (x + 1)(x - 1)(x^2 - x + 1)(x^2 + x + 1)$$
> expand (eq) ;

$$x^6 - 1$$

Разложение многочлена на множители осуществляется командой **factor (eq)**. Пример:

```
> p:=x^5-x^4-7*x^3+x^2+6*x;
      p := x5 - x4 - 7x3 + x2 + 6x
> factor(p);
      x(x-1)(x-3)(x+2)(x+1)
```

Команда **expand** может иметь дополнительный параметр, позволяющий при раскрытии скобок оставлять определенное выражение без изменений. Например, пусть требуется каждое слагаемое выражения $\ln x + e^x - y^2$ умножить на выражение $(x+a)$.

Тогда в командной строке следует написать:

```
> expand((x+a)*(ln(x)+exp(x)-y^2), (x+a));
      (x+a)ln x + (x+a)ex - (x+a)y2
```

Дробь можно привести к нормальному виду с помощью команды **normal(eq)**. Например:

```
> f:=(a^4-b^4)/((a^2+b^2)*a*b);
      f :=  $\frac{a^4 - b^4}{(a^2 + b^2)ab}$ 
```

```
> normal(f);
       $\frac{a^2 - b^2}{ab}$ 
```

Упрощение выражений осуществляется командой **simplify(eq)**. Пример:

```
> eq:=(cos(x)-sin(x))*(cos(x)+sin(x));
> simplify(eq);
      2cos(x)2 - 1
```

Приведение подобных членов в выражении осуществляется командой **collect(exp,var)**, где **exp** – выражение, **var** – имя переменной, относительно которой следует собирать подобные. В команде **simplify** в качестве параметров можно указать, какие выражения преобразовывать. Например, при указании **simplify(eq, trig)** будет производиться упрощение при использовании большого числа тригонометрических соотношений. Стандартные параметры имеют названия: **power** – для степенных преобразований; **radical** или **sqrt** – для преобразования корней; **exp** – преобразование экспонент; **ln** – преобразование логарифмов. Использование параметров намного увеличивает эффективность команды **simplify**.

Объединить показатели степенных функций или понизить степень тригонометрических функций можно при помощи команды **combine(eq, param)**, где **eq** – выражение, **param** – параметры, указывающие, какой тип функций преобразовать, например, **trig** – для тригонометрических, **power** – для степенных. Пример:

```
> combine(4*sin(x)^3, trig);  
- sin(3x) + 3 sin(x)
```

Для упрощения выражений, содержащих не только квадратные корни, но и корни других степеней, лучше использовать команду **radnormal(eq)**. Пример:

```
> sqrt(3+sqrt(3)+(10+6*sqrt(3))^(1/3))=  
radnormal(sqrt(3+sqrt(3)+(10+6*sqrt(3))^(1/3)));  

$$\sqrt{3+\sqrt{3}+(10+6\sqrt{3})^{1/3}} = 1+\sqrt{3}$$

```

С помощью команды **convert(exp, param)**, где **exp** – выражение, которое будет преобразовано в указанный тип **param**. В частности, можно преобразовать выражение, содержащее $\sin x$ и $\cos x$, в выражение, содержащее только $\operatorname{tg} x$, если указать в качестве параметра **tan**, или, наоборот, $\operatorname{tg} x$, $\operatorname{ctg} x$ можно перевести в $\sin x$ и $\cos x$, если в параметрах указать **sincos**.

Вообще, команда **convert** имеет более широкое назначение. Она осуществляет преобразование выражения одного типа в другой. Например: **convert(list, vector)** – преобразование некоторого списка **list** в вектор с теми же элементами; **convert(expr, string)** – преобразование математического выражения в его текстовую запись. Для вызова подробной информации о назначении параметров команды **convert** следует обратиться к справочной системе, набрав **convert[termin]**.

Если вы забыли параметры какой-либо команды, то можно воспользоваться справочной системой *Maple*. Для вызова справки по конкретной команде, следует выделить набранное имя этой команды и нажать клавишу F1. Если команда набрана правильно, то появится описание этой команды (в большинстве версий *Maple* помощь на английском языке).

Задание 4.

1. Перейдите в текстовый режим и наберите «Задание №4». После не забудьте перейти в режим командной строки. Перед выполнением

Методы решения математических задач в Maple

каждого пункта этого задания обязательно набирайте команду обновления **restart**;

2. Разложить полином на множители $p = x^3 + 4x^2 + 2x - 4$. Для этого наберите в командной строке:
> **factor(x^3+4*x^2+2*x-4)** ;

После нажатия клавиши *Enter* должно получиться $(x+2)(x^2+2x-2)$.

3. Упростить выражение $\frac{1 + \sin 2x + \cos 2x}{1 + \sin 2x - \cos 2x}$. Наберите:
> **eq:=(1+sin(2*x)+cos(2*x))/(1+sin(2*x)-cos(2*x))** ;
> **convert(eq, tan)** ;
> **eq=normal(%)** ;

$$\frac{1 + \sin(2x) + \cos(2x)}{1 + \sin(2x) - \cos(2x)} = \frac{1}{\tan(x)}$$

4. Упростить выражение $3(\sin^4 x + \cos^4 x) - 2(\sin^6 x + \cos^6 x)$. Для этого наберите:

> **eq:=3*(sin(x)^4+cos(x)^4)-2*(sin(x)^6+cos(x)^6)** ;
> **eq=combine(eq, trig)** ;

$$3\sin(x)^4 + 3\cos(x)^4 - 2\sin(x)^6 + \cos(x)^6 = 1$$

5. Выполните все контрольные задания. Перед их выполнением не забудьте набрать в текстовом режиме «Контрольные задания». Результаты выполнения заданий покажите преподавателю.
6. Сохраните файл со всеми выполненными заданиями на диск.
7. Ответьте на все контрольные вопросы.

Контрольные задания.

1. Вычислить: $(-1+i)^5$.
2. Вычислить: $e^{i\pi/2}$.
3. Вычислить точное и значение выражения: $\arctg 3 - \arcsin \frac{\sqrt{5}}{5}$.
4. Записать формулы: $\omega(k) = \alpha k^2 + \beta k^4$; $\xi = ae^{-\gamma r} \cos(\omega t + \varphi)$.
5. Разложить на множители полином $p = x^3 - 4x^2 + 5x - 2$.
6. Упростить выражение $\sin^2 3x - \sin^2 2x - \sin 5x \sin x$.

Контрольные вопросы.

1. Что такое *Maple* и для чего он предназначен?
2. Опишите основные элементы окна *Maple*.
3. На какие условные части делится рабочее поле *Maple* и что в этих частях отображается?
4. Как перевести командную строку в текстовую и наоборот?
5. В каком режиме проходит сеанс работы в *Maple*?
6. Перечислите пункты основного меню *Maple* и их назначение.
7. Какое стандартное расширение присваивается файлу рабочего листа *Maple*?
8. Как представляются в *Maple* основные математические константы?
9. Опишите виды представления рационального числа в *Maple*.
10. Как получить приближенное значение рационального числа?
11. Какими разделительными знаками заканчиваются команды в *Maple* и чем они отличаются?
12. Какой командой осуществляется вызов библиотеки подпрограмм?
13. Объясните назначение команд **factor**, **expand**, **normal**, **simplify**, **combine**, **convert**.

II. Функции в *Maple*. Операции оценивания. Решение уравнений и неравенств

1. Способы задания функций. Замена переменных.
2. Операции оценивания.
3. Решение уравнений.
4. Решение неравенств.

§1. Способы задания функций. Замена переменных

В *Maple* имеется несколько способов представления функции.

Способ 1. Определение функции с помощью оператора присваивания (**:=**): какому-то выражению присваивается имя, например:

```
> f := sin(x) + cos(x) ;
```

$$f := \sin(x) + \cos(x)$$

Если задать конкретное значение переменной x , то получится значение функции f для этого x . Например, если продолжить предыдущий пример и вычислить значение \mathbf{f} при $x = \pi/4$, то следует записать:

```
> x:=Pi/4;
```

$$x := \frac{\pi}{4}$$

```
> f;
```

$$\sqrt{2}$$

После выполнения этих команд переменная x имеет заданное значение $\pi/4$.

Чтобы насовсем не присваивать переменной конкретного значения, удобнее использовать команду подстановки **subs** (**{x1=a1, x2=a2, ..., }**, **f**), где в фигурных скобках указываются переменные \mathbf{x}_i и их новые значения \mathbf{a}_i ($i=1,2,\dots$), которые следует подставить в функцию \mathbf{f} . Например:

```
> f:=x*exp(-t);
```

$$f := xe^{(-t)}$$

```
> subs({x=2, t=1}, f);
```

$$2e^{(-1)}$$

Все вычисления в *Maple* по умолчанию производятся символьно, то есть результат будет содержать в явном виде иррациональные константы, такие как, e , π и другие. Чтобы получить приближенное значение в виде числа с плавающей запятой, следует использовать команду **evalf** (**expr**, **t**), где **expr** – выражение, **t** – точность, выраженная в числах после запятой. Например, в продолжение предыдущего примера, вычислим полученное значение функции приближенно:

```
> evalf(%);
```

$$.7357588824$$

Здесь использован символ (%) для вызова предыдущей команды.

Способ 2. Определение функции с помощью функционального оператора, который ставит в соответствие набору переменных ($\mathbf{x}_1, \mathbf{x}_2, \dots$) одно или несколько выражений ($\mathbf{f}_1, \mathbf{f}_2, \dots$). Например, определение функции двух переменных с помощью функционального оператора выглядит следующим образом:

```
> f:=(x, y) -> sin(x+y);
```

$$f := \sin(x + y)$$

Обращение к этой функции осуществляется наиболее привычным в математике способом, когда в скобках вместо аргументов функции указываются конкретные значения переменных. В продолжение предыдущего примера вычисляется значение функции:

```
> f(Pi/2, 0);
```

1

Способ 3. С помощью команды **unapply(expr, x1, x2, ...)**, где **expr** – выражение, **x1, x2, ...** – набор переменных, от которых оно зависит, можно преобразовать выражение **expr** в функциональный оператор. Например:

```
> f:=unapply(x^2+y^2, x, y);
```

$$f := (x, y) \rightarrow x^2 + y^2$$

```
> f(-7, 5);
```

74

В *Maple* имеется возможность определения неэлементарных функций вида

$$f(x) = \begin{cases} f_1(x), & x < a_1 \\ f_2(x), & a_1 < x < a_2 \\ \dots\dots\dots \\ f_n(x), & x > a_n \end{cases}$$

посредством команды

```
> piecewise(cond_1, f1, cond_2, f2, ...).
```

Например, функция

$$f(x) = \begin{cases} 0, & x < 0 \\ x, & 0 \leq x < 1 \\ \sin x, & x \geq 1 \end{cases}$$

записывается следующим образом:

```
> f:=piecewise(x<0, 0, 0<=x and x<1, x, x>=1, sin(x));
```

$$f := \begin{cases} 0 & x < 0 \\ x & -x \leq 0 \text{ and } x - 1 < 0 \\ \sin x & 1 \leq x \end{cases}$$

Задание 1.

Не забудьте, что выполнение всех последующих заданий должно начинаться с текстовой строки, содержащей «Задание №», где № – номер задания. Также помните, что для правильности вычислений перед выполнением каждого пункта задания следует выполнять команду **restart**. Перед выполнением контрольных заданий следует набирать в текстовом режиме «Контрольные задания». Эти правила оформления относятся ко всем лабораторным работам.

1. Запустите *Maple*. Переведите первую строку в текстовую и наберите в ней: «Лабораторная работа №2». Нажмите *Enter*. Строкой ниже наберите: «Выполнил студент ...» и свою фамилию, а на следующей строке наберите: «Задание №1».

2. Определите функцию $f = \sqrt{1 - x^2 - y^2}$ и перейдите в ней к полярным координатам $x = \rho \cos \phi$, $y = \rho \sin \phi$. Упростите полученное выражение. Для этого наберите:

> f:=sqrt(1-x^2-y^2);

$$f = \sqrt{1 - x^2 - y^2}$$

> f:=subs({x=rho*cos(phi), y=rho*sin(phi)}, f);

$$f = \sqrt{1 - \rho^2 \cos(\phi)^2 - \rho^2 \sin(\phi)^2}$$

> f:=simplify(%);

$$f = \sqrt{1 - \rho^2}$$

3. Определите функцию $f(x) = \begin{cases} x, & x < -1 \\ -x^2, & -1 \leq x < 1 \\ -x, & x \geq 1 \end{cases}$ и прибавьте к ней x .

Для этого наберите:

> f:=piecewise(x<-1, x, -1<=x and x<1, -x^2, x>=1, -x);

$$f := \begin{cases} x & x < -1 \\ -x^2 & -1 - x \leq 0 \text{ and } x - 1 < 0 \\ -x & 1 \leq x \end{cases}$$

> %+x: simplify(%);

$$\begin{cases} 2x & x < -1 \\ x - x^2 & x \leq 1 \\ 0 & 1 < x \end{cases}$$

§2. Операции оценивания

Оценивание вещественных выражений.

В *Maple* имеются следующие команды оценивания вещественных выражений:

frac (expr) – вычисление дробной части выражения **expr**;

trunc (expr) – вычисление целой части выражения **expr**;

round (expr) – округление выражения **expr**;

Оценивание комплексных выражений.

Вещественную и мнимую части комплексного выражения $z=x+iy$ можно найти с помощью команд **Re (z)** и **Im (z)**. Например:

```
> z:=3+I*2:
```

```
> Re (z) ; Im (z) ;
```

3, 2

Если $z=x+iy$, то комплексно сопряженное ему выражение $w=z^*=x-iy$ можно найти с помощью команды **conjugate (z)**.

Продолжение предыдущего примера:

```
w:=conjugate (z) ;
```

$w:=3-2 I$

Модуль и аргумент комплексного выражения **z** можно найти с помощью команды **polar (z)**, которую необходимо предварительно вызвать из стандартной библиотеки командой **readlib**. Например:

```
> readlib (polar) : polar (I) ;
```

$\text{polar} \left(1, \frac{1}{2} \pi \right)$

В строке вывода в скобках через запятую указаны модуль числа i , равный единице и его аргумент, равный $\pi/2$.

Если комплексное выражение очень сложное или содержит параметры, то команды **Re (z)** и **Im (z)** не дают требуемого результата. Получить вещественную и мнимую части комплексного выражения **z** можно, если использовать команду преобразования комплексных выражений **evalc (z)**. Например:

```
> z:=ln (1-I*sqrt (3)) ^2 ;
```

$$z := \ln(1 - I\sqrt{3})^2$$

$$> \text{evalc}(\text{Re}(z)); \text{evalc}(\text{Im}(z));$$

$$\frac{1}{4}\ln(4)^2 - \frac{1}{9}\pi^2$$

$$-\frac{1}{3}\ln(4)\pi$$

Задание 2.

1. Дано число $a=57/13$. Найти его целую часть x и дробную часть y и убедиться, что $a=x+y$. Наберите:

> a:=57/13:

> y:=frac(a);

$$\frac{5}{13}$$

> x:=trunc(a);

$$4$$

> x+y;

$$\frac{57}{13}$$

2. Дано комплексное число $z = \frac{2-3i}{1+4i} + i^6$. Найти его вещественную и мнимую части, а затем комплексно сопряженное ему число w и убедиться, что $w+z=2\text{Re}(z)$.

В командной строке наберите:

> z:=(2-3*I)/(1+4*I)+I^6:

> Re(z); Im(z);

$$-\frac{27}{17}$$

$$-\frac{11}{17}$$

> w:=conjugate(z);

$$w := -\frac{27}{17} + \frac{11}{17}I$$

> z+w;

$$-\frac{54}{17}$$

3. Найти модуль и аргумент комплексного числа $z = -1 - i\sqrt{3}$ и вычислить z^4 . Наберите:

```
> z:=-1-I*sqrt(3);
> readlib(polar): polar(z);
      polar(2,-2/3*pi)
```

Чему равен модуль и аргумент этого числа?

```
> evalc(z^4);
      -8-8*sqrt(3)*I
```

§3. Решение уравнений

Решение обыкновенных уравнений.

Для решения уравнений в *Maple* существует универсальная команда **solve(eq, x)**, где **eq** – уравнение, **x** – переменная, относительно которой уравнение надо разрешить. В результате выполнения этой команды в строке вывода появится выражение, которое является решением данного уравнения. Например:

```
> solve(a*x+b=c, x);
      b-c
      -
      a
```

Если уравнение имеет несколько решений, которые вам понадобятся для дальнейших расчетов, то команде **solve** следует присвоить какое-нибудь имя **name**. Обращение к какому-либо **k**-ому решению данного уравнения производится указанием его имени с номером решения **k** в квадратных скобках: **name[k]**. Например:

```
> x:=solve(x^2-a=0, x);
      x := -sqrt(a), sqrt(a)
> x[1];
      -sqrt(a)
> x[2];
      sqrt(a)
> x[1]+x[2];
      0
```

Решение систем уравнений.

Системы уравнений решаются с помощью такой же команды **solve** (**{eq1, eq2, ...}**, **{x1, x2, ...}**), только теперь в параметрах команды следует указывать в первых фигурных скобках через запятую уравнения, а во вторых фигурных скобках перечисляются через запятую переменные, относительно которых требуется решить систему. Если вам будет необходимо для дальнейших вычислений использовать полученные решения уравнений, то команде **solve** следует присвоить какое-нибудь имя **name**. Затем выполняется присвоения команда **assign(name)**. После этого над решениями можно будет производить математические операции. Например:

```
> s:=solve({a*x-y=1, 5*x+a*y=1}, {x, y});
```

$$s:=\left\{x=\frac{a+1}{5+a^2}, y=\frac{a-5}{5+a^2}\right\}$$

```
> assign(s); simplify(x-y);
```

$$\frac{6}{5+a^2}$$

Численное решение уравнений.

Для численного решения уравнений, в тех случаях, когда трансцендентные уравнения не имеют аналитических решений, используется специальная команда **fsolve**(**eq, x**), параметры которой такие же, как и команды **solve**. Например:

```
> x:=fsolve(cos(x)=x, x);
```

$$x:=-.7390851332$$

Решение рекуррентных и функциональных уравнений.

Команда **rsolve**(**eq, f**) позволяет решить рекуррентное уравнение **eq** для целой функции **f**. Можно задать некоторое начальное условие для функции **f(n)**, тогда получится частное решение данного рекуррентного уравнения. Например:

```
> eq:=2*f(n)=3*f(n-1)-f(n-2);
```

$$eq:=2f(n)=3f(n-1)-f(n-2)$$

```
> rsolve({eq, f(1)=0, f(2)=1}, f);
```

$$2-4\left(\frac{1}{2}\right)^n$$

Универсальная команда **solve** позволяет решать функциональные уравнения, например:

```
> F:=solve (f (x) ^2-3*f (x) +2*x, f) ;
      F:=proc(x) RootOf(_Z^2 - 3*_Z + 2*x) end
```

В результате получается решение в неявном виде. Однако *Maple* может работать с такими решениями. Неявное решение функционального уравнения можно попытаться преобразовать в какую-либо элементарную функцию с помощью команды **convert**. Продолжая приведенный выше пример, можно получить решение в явном виде:

```
> f:=convert (F (x) , radical) ;
      f := 3/2 + 1/2*sqrt(9-8x)
```

Решение тригонометрических уравнений.

Команда **solve**, примененная для решения тригонометрического уравнения, выдает только главные решения, то есть решения в интервале $[0, 2\pi]$. Для того, чтобы получить все решения, следует предварительно ввести дополнительную команду **_EnvAllSolutions:=true**. Например:

```
> _EnvAllSolutions:=true;
> solve (sin (x)=cos (x) , x) ;
      1/4*pi + pi*_Z ~
```

В *Maple* символ $_Z$ обозначает константу целого типа, поэтому решение данного уравнения в привычной форме имеет вид $x := \pi/4 + \pi n$, где n – целые числа.

Решение трансцендентных уравнений.

При решении трансцендентных уравнений для получения решения в явном виде перед командой **solve** следует ввести дополнительную команду **_EnvExplicit:=true**. Пример решения сложной системы трансцендентных уравнений и упрощения вида решений:

```
> eq:={ 7*3^x-3*2^(z+y-x+2)=15, 2*3^(x+1)+
3*2^(z+y-x)=66, ln(x+y+z)-3*ln(x)-ln(y*z)=-ln(4) } :
> _EnvExplicit:=true:
> s:=solve (eq, {x,y,z}) :
> simplify (s[1]) ; simplify (s[2]) ;
      {x=2, y=3, z=1}, {x=2, y=1, z=3}
```

Задание 3.

1. Найти все решения системы уравнений $\begin{cases} x^2 - y^2 = 1, \\ x^2 + xy = 2. \end{cases}$

Наберите:

```
> eq:={x^2-y^2=1,x^2+x*y=2};
```

```
> _EnvExplicit:=true:
```

```
> s:=solve(eq,{x,y});
```

$$s := \{x = \frac{2}{3}\sqrt{3}, y = \frac{1}{3}\sqrt{3}\}, \{x = -\frac{2}{3}\sqrt{3}, y = -\frac{1}{3}\sqrt{3}\}$$

Теперь найдите сумму двух наборов решений. Наберите:

```
> x1:=subs(s[1],x): y1:=subs(s[1],y):
```

```
  x2:=subs(s[2],x): y2:=subs(s[2],y):
```

```
> x1+x2; y1+y2;
```

Чему равны эти суммы решений?

2. Численно решите уравнение $x^2 = \cos(x)$. Наберите:

```
> x=fsolve(x^2=cos(x),x);
```

$$x = .8241323123$$

3. Найдите функцию $f(x)$, удовлетворяющую уравнению $f^2(x) - 2f(x) = x$. Наберите:

```
> F:=solve(f(x)^2-2*f(x)=x,f);
```

```
  F:=proc(x) RootOf(_Z^2-2*_Z-x) end
```

```
> f:=convert(F(x), radical);
```

$$f := 1 + \sqrt{1+x}$$

4. Найдите все решения уравнения $5\sin x + 12\cos x = 13$. Наберите:

```
> _EnvAllSolutions:=true:
```

```
> solve(5*sin(x)+12*cos(x)=13,x);
```

$$\arctan\left(\frac{5}{12}\right) + 2\pi_Z \sim$$

§4. Решение неравенств

Решение простых неравенств.

Команда **solve** применяется также для решения неравенств. Решение неравенства выдается в виде интервала изменения искомой переменной. В том случае, если решение неравенства полуось, то в поле вывода появляется конструкция вида $\text{RealRange}(-\infty, \text{Open}(a))$,

которая означает, что $x \in (-\infty, a)$, a – некоторое число. Слово `Open` означает, что интервал с открытой границей. Если этого слова нет, то соответствующая граница интервала включена во множество решений. Например:

```
> s:=solve(sqrt(x+3)<sqrt(x-1)+sqrt(x-2),x);
> convert(s,radical);
```

$$\text{RealRange}\left(\text{Open}\left(\frac{2}{3}\sqrt{21}\right),\infty\right)$$

Если вы хотите получить решение неравенства не в виде интервального множества типа $x \in (a, b)$, а в виде ограничений для искомой переменной типа $a < x$, $x < b$, то переменную, относительно которой следует разрешить неравенство, следует указывать в фигурных скобках. Например:

```
> solve(1-1/2*ln(x)>2,{x});
{0 < x, x < e^{(-2)}}
```

Решение систем неравенств.

С помощью команды `solve` можно также решить систему неравенств. Например:

```
> solve({x+y>=2,x-2*y<=1,x-y>=0,x-2*y>=1},{x,y});
{x = 1 + 2y, 1/3 <= y}
```

Задание 4.

1. Решите неравенство $13x^3 - 25x^2 - x^4 - 129x + 270 > 0$. Наберите:

```
> solve(13*x^3-25*x^2-x^4-129*x+270>0,x);
```

```
RealRange(Open(-3),Open(2)),RealRange(Open(5),Open(9))
```

Запишите этот результат в аналитическом виде. Получите решение этого неравенства в виде ограничений для искомой переменной. Прodelайте это самостоятельно.

2. Решите неравенство $e^{(2x+3)} < 1$. Наберите:

```
> solve(exp(2*x+3)<1,x);
```

$$\text{RealRange}\left(-\infty,\text{Open}\left(-\frac{2}{3}\right)\right)$$

Теперь получите самостоятельно решение этого неравенства в виде ограничений для искомой переменной.

Методы решения математических задач в Maple

3. Выполните все контрольные задания. Перед их выполнением наберите в текстовом режиме «Контрольные задания». Результаты выполнения заданий покажите преподавателю. Сохраните файл со всеми выполненными заданиями на диск. Ответьте на все контрольные вопросы.

Контрольные задания.

1. Дано комплексное число $z = (2e^{i\pi/6})^5$. Найти его вещественную и мнимые части, алгебраическую форму, модуль и аргумент.
2. Записать функцию $f(x, y) = \left(\frac{\operatorname{arctg}(x+y)}{\operatorname{arctg}(x-y)} \right)^2$ в виде функционального оператора и вычислите ее значения при $x=1, y=0$ и при $x = (1 + \sqrt{3})/2, y = (1 - \sqrt{3})/2$.
3. Записать функцию $f(x, y) = \frac{x^3 y^2 - x^2 y^3}{(xy)^5}$ с помощью оператора присваивания и вычислите ее значение при $x=a, y=1/a$, используя команду подстановки **subs**.
4. Найти все точные решения системы $\begin{cases} x^2 - 5xy + 6y^2 = 0, \\ x^2 + y^2 = 10. \end{cases}$ в аналитическом виде.
5. Найти все решения тригонометрического уравнения $\sin^4 x - \cos^4 x = 1/2$.
6. Найти численное решение уравнения $e^x = 2(1-x)^2$.
7. Решить неравенство $2\ln^2 x - \ln x < 1$.

Контрольные вопросы.

1. Опишите способы задания функций в Maple.
2. Какие операции оценивания производятся в Maple с действительными выражениями?
3. Для чего предназначена команда **evalf**?
4. С помощью каких команд можно найти вещественную и мнимую части комплексного выражения, а также его модуль и аргумент, и комплексно сопряженное ему число? Какую роль выполняет команда **evalc**?
5. Для чего предназначена команда **solve**?

6. Какие команды используются для численного решения уравнений и для решения рекуррентных уравнений?
7. Какие дополнительные команды следует ввести для того, чтобы получить точное решение уравнения, все решения уравнения?
8. В каком виде выдается решение неравенства? Как отличить в строке вывода закрытый интервал от открытого?

III. Построение графиков

1. Двумерные графики.
2. Трехмерные графики. Анимация.

§1. Двумерные графики

Команда `plot` и ее параметры.

Для построения графиков функции $f(x)$ одной переменной (в интервале $a \leq x \leq b$ по оси Ox и в интервале $c \leq y \leq d$ по оси Oy) используется команда `plot(f(x), x=a..b, y=c..d, parameters)`, где `parameters` – параметры управления изображением. Если их не указывать, то будут использованы установки по умолчанию. Настройка изображения также может осуществляться с панели инструментов.

Основные параметры команды `plot`:

- 1) `title="text"`, где `text` – заголовок рисунка (текст можно оставлять без кавычек, если он содержит только латинские буквы без пробелов).
- 2) `coords=polar` – установка полярных координат (по умолчанию установлены декартовы).
- 3) `axes` – установка типа координатных осей: `axes=NORMAL` – обычные оси; `axes=BOXED` – график в рамке со шкалой; `axes=FRAME` – оси с центром в левом нижнем углу рисунка; `axes=NONE` – без осей.
- 4) `scaling` – установка масштаба рисунка: `scaling=CONSTRAINED` – одинаковый масштаб по осям; `scaling=UNCONSTRAINED` – график масштабируется по размерам окна.
- 5) `style=LINE (POINT)` – вывод линиями (или точками).
- 6) `numpoints=n` – число вычисляемых точек графика (по умолчанию `n=49`).

- 7) **color** – установка цвета линии: английское название цвета, например, **yellow** – желтый и т.д.
- 8) **xtickmarks=nx** и **ytickmarks=ny** – число меток по оси Ox и оси Oy , соответственно.
- 9) **thickness=n**, где **n=1,2,3...** - толщина линии (по умолчанию **n=1**).
- 10) **linestyle=n** – тип линии: непрерывная, пунктирная и т.д. (**n=1** – непрерывная, установлено по умолчанию).
- 11) **symbol=s** – тип символа, которым помечают точки: **BOX**, **CROSS**, **CIRCLE**, **POINT**, **DIAMOND**.
- 12) **font=[f,style,size]** – установка типа шрифта для вывода текста: **f** задает название шрифтов: **TIMES**, **COURIER**, **HELVETICA**, **SYMBOL**; **style** задает стиль шрифта: **BOLD**, **ITALIC**, **UNDERLINE**; **size** – размер шрифта в pt.
- 13) **labels=[tx,ty]** – надписи по осям координат: **tx** – по оси Ox и **ty** – по оси Oy .
- 14) **discont=true** – указание для построения бесконечных разрывов.


С помощью команды **plot** можно строить помимо графиков функций **y=f(x)**, заданной явно, также графики функций, заданных параметрически **y=y(t)**, **x=x(t)**, если записать команду **plot([y=y(t), x=x(t), t=a..b], parameters)**.

Задание 1.1.

1. Построить график $y = \frac{\sin x}{x}$ жирной линией в интервале от -4π до


4π. Наберите:

```
> plot(sin(x)/x, x=-4*Pi..4*Pi, labels=[x,y],
  labelfont=[TIMES,ITALIC,12], thickness=2);
```



2. Построить график разрывной функции $y = \frac{x}{x^2 - 1}$.


`> plot(x/(x^2-1), x=-3..3, y=-3..3, color=magenta);`



Замечание: на рисунке автоматически появляются вертикальные асимптоты.


3. Построить график параметрической кривой $y = \sin 2t$, $x = \cos 3t$, $0 \leq t \leq 2\pi$ в рамке. Наберите:

`> plot([sin(2*t), cos(3*t), t=0..2*Pi], axes=BOXED, color=blue);`




4. Построить в полярных координатах график кардиоиды $\rho = 1 + \cos \varphi$ с названием. Наберите:

`> plot(1+cos(x), x=0..2*Pi, title="Cardioida", coords=polar, color=coral, thickness=2);`



5. Построить два графика на одном рисунке: график функции $y = \ln(3x - 1)$ и касательную к нему $y = \frac{3}{2}x - \ln 2$. Наберите:

```
> plot([ln(3*x-1), 3*x/2-ln(2)], x=0..6,
scaling=CONSTRAINED, color=[violet,gold],
linestyle=[1,2], thickness=[3,2]);
```



Построение графика функции, заданной неявно.

Функция задана неявно, если она задана уравнением $F(x, y) = 0$.

Для построения графика неявной функции используется команда `implicitplot` из графического пакета `plots`: `implicitplot(F(x,y)=0, x=x1..x2, y=y1..y2)`.

Вывод текстовых комментариев на рисунок.

В пакете `plots` имеется команда `textplot` для вывода текстовых комментариев на рисунок: `textplot([x0,y0,'text'], options)`, где x_0 , y_0 – координаты точки, с которой начинается вывод текста 'text'.

Вывод нескольких графических объектов на один рисунок.

Часто бывает необходимо совместить на одном рисунке несколько графических объектов, полученных при помощи различных команд, например, добавить графику, нарисованную командой **plot**, текстовые надписи, полученные командой **textplot**. Для этого результат действия команды присваивается некоторой переменной:

```
> p:=plot(...): t:=textplot(...):
```

При этом на экран вывод не производится. Для вывода графических изображений необходимо выполнить команду из пакета **plots**:

```
> with(plots): display([p,t], options).
```

Построение двумерной области, заданной неравенствами.

Если необходимо построить двумерную область, заданную системой неравенств $f_1(x, y) > c_1, f_2(x, y) > c_2, \dots, f_n(x, y) > c_n$, то для этого можно использовать команду **inequal** из пакета **plots**. В команде **inequals({f1(x,y)>c1,...,fn(x,y)>cn}, x=x1..x2, y=y1..y2, options)** в фигурных скобках указывается система неравенств, определяющих область, затем размеры координатных осей и параметры. Параметры регулируют цвета открытых и закрытых границ, цвета внешней и внутренней областей, а также толщину линий границ:


- **optionsfeasible=(color=red)** – установка цвета внутренней области;
- **optionsexcluded=(color=yellow)** – установка цвета внешней области;
- **optionsopen(color=blue, thickness=2)** – установка цвета и толщины линии открытой границы;
- **optionsclosed(color=green, thickness=3)** – установка цвета и толщины линии закрытой границы.

Задание 1.2.

1. Построить график неявной функции (гиперболы): $\frac{x^2}{4} - \frac{y^2}{2} = 16$.


```
> with(plots):
```

```
> implicitplot(x^2/4-y^2/2=16, x=-20..20,  
y=-16..16,  
color=green, thickness=2);
```




2. Построить на одном рисунке графики астроиды $x = 4\cos^3 t$, $y = 2\sin^3 t$ ($0 \leq t \leq 2\pi$) вписанной в эллипс $\frac{x^2}{16} + \frac{y^2}{4} = 1$. Выведите название линий **Astroida** и **Ellips** жирным шрифтом вместе с его уравнением курсивом. Для этого наберите следующие строки:

```
> with(plots):
> eq:=x^2/16+y^2/4=1:
> el:=implicitplot(eq, x=-4..4, y=-2..2,
scaling=CONSTRAINED,
color=green, thickness=3):
> as:=plot([4*cos(t)^3,2*sin(t)^3, t=0..2*Pi],
color=blue, scaling=CONSTRAINED, thickness=2):
> eq1:=convert(eq,string):
> t1:=textplot([1.5,2.5,eq1], font=[TIMES,
ITALIC, 10], align=RIGHT):
> t2:=textplot([0.2,2.5,"Ellips:"], font=[TIMES,
BOLD,10], align=RIGHT):
> t3:=textplot([1.8,0.4,Astroida], font=[TIMES,
BOLD,10], align=LEFT):
> display([as,el,t1,t2,t3]);
```



3. Построить область, ограниченную линиями: $x + y > 0$, $x - y \leq 1$, $y = 2$.

```
> with(plots) :
> inequal({x+y>0, x-y<=1, y=2}, x=-3..3, y=-3..3,
optionsfeasible=(color=red),
optionsopen=(color=blue,thickness=2),
optionsclosed=(color=green, thickness=3),
optionsexcluded=(color=yellow) );
```



§2. Трехмерные графики. Анимация

График поверхности, заданной явной функцией.

График функции $z = f(x, y)$ можно нарисовать, используя команду `plot3d(f(x,y), x=x1..x2, y=y1..y2, options)`. Параметры этой команды частично совпадают с параметрами команды `plot`. К часто используемым параметрам команды `plot3d` относится `light=[angl1, angl2, c1, c2, c3]` – задание подсветки поверхности, создаваемой источником света из точки со сферическими координатами (`angl1, angl2`). Цвет определяется долями красного (`c1`), зеленого (`c2`) и синего (`c3`) цветов, которые находятся в интервале $[0,1]$. Параметр `style=opt` задает стиль рисунка: **POINT** – точки, **LINE** – линии, **HIDDEN** – сетка с удалением невидимых линий, **PATCH** – заполнитель (установлен по умолчанию), **WIREFRAME** – сетка с выводом невидимых линий, **CONTOUR** – линии уровня, **PATCHCONTOUR** – заполнитель и линии уровня. Параметр `shading=opt` задает функцию интенсивности заполнителя, его значение равно `huz` – по умолчанию, **NONE** – без раскраски.

График поверхности, заданной параметрически.

Если требуется построить поверхность, заданную параметрически: $x=x(u,v)$, $y=y(u,v)$, $z=z(u,v)$, то эти функции перечисляются в квадратных скобках в команде: `plot3d([x(u,v), y(u,v), z(u,v)], u=u1..u2, v=v1..v2)`.

График поверхности, заданной неявно.

Трёхмерный график поверхности, заданной неявно уравнением $F(x,y,z)=c$, строится с помощью команды пакета `plot`: `implicitplot3d(F(x,y,z)=c, x=x1..x2, y=y1..y2, z=z1..z2)`, где указывается уравнение поверхности $F(x,y,z)=c$ и размеры рисунка по координатным осям.

График пространственных кривых.

В пакете `plot` имеется команда `spacecurve` для построения пространственной кривой, заданной параметрически: $x = x(t), y = y(t), z = z(t)$. Параметры команды:

> `spacecurve([x(t), y(t), z(t)], t=t1..t2)`,

где переменная **t** изменяется от **t1** до **t2**.

Анимация.

Maple позволяет выводить на экран движущиеся изображения с помощью команд `animate` (двумерные) и `animate3d` (трехмерные) из пакета `plot`. Среди параметров команды `animate3d` есть `frames` – число кадров анимации (по умолчанию `frames=8`).

Трёхмерные изображения удобнее настраивать не при помощи опций команды `plot3d`, а используя контекстное меню программы. Для этого следует щелкнуть правой кнопкой мыши по изображению. Тогда появится контекстное меню настройки изображения. Команды этого меню позволяют изменять цвет изображения, режимы подсветки, устанавливать нужный тип осей, тип линий и управлять движущимся изображением.


Контекстное меню настройки изображения:



Задание 2.

1. Выполнить построение двух поверхностей $z = x \sin 2y + y \cos 3x$ и $z = \sqrt{x^2 + y^2} - 7$ в пределах $(x, y) \in [-\pi, \pi]$. Установите переменный цвет поверхностей как функцию $x + y$.

```
> plot3d({x*sin(2*y)+y*cos(3*x), sqrt(x^2+y^2)-7},
x=-Pi..Pi, y=-Pi..Pi, grid=[30,30], axes=FRAMED,
color=x+y);
```




2. Построить поверхность

$$z = \frac{1}{x^2 + y^2} + \frac{0,2}{(x+1,2)^2 + (y-1,5)^2} + \frac{0,3}{(x-0,9)^2 + (y+1,1)^2} \text{ вместе}$$

с линиями уровня:

```
> plot3d(1/(x^2+y^2)+0.2/((x+1.2)^2+(y-1.5)^2)+
0.3/((x-0.9)^2+(y+1.1)^2), x=-2..2, y=-2..2.5,
view=[-2..2, -2..2.5, 0..6], grid=[60,60],
shading=NONE, light=[100,30,1,1,1], axes=NONE,
orientation=[65,20], style=PATCHCONTOUR);
```



3. Построить примерную форму электронного облака атома. Форма электронного облака определяется двумя квантовыми числами: число l – определяет тип орбитали, число m – определяет тип магнитный момент электрона. При $m=0$ форма электронного облака задается полиномами Лежандра первого рода:

$$P(x) = \frac{1}{2^n n!} \frac{d^n}{dx^n} (x^2 - 1)^n . \text{ Следует построить параметрически}$$

заданную поверхность: $x(\theta, \varphi) = Y(\varphi) \sin \varphi \cos \theta$,
 $y(\theta, \varphi) = Y(\varphi) \sin \varphi \sin \theta$, $z(\theta, \varphi) = Y(\varphi) \cos \varphi$, где

$$Y(\varphi) = \sqrt{\frac{2l+1}{4\pi}} P(\cos \varphi) . \text{ Вначале положите } l=3 . \text{ Наберите}$$

команды:


```
> l:=3;
> P:=(x,n)->1/(2^n*n!)*diff((x^2-1)^n,x$n);
> Y:=(phi)->abs(sqrt((2*l+1)/(4*Pi))*
  subs(x=cos(phi),P(x,l)));
> X0:=Y(phi)*sin(phi)*cos(theta);
> Y0:=Y(phi)*sin(phi)*sin(theta);
> Z0:=Y(phi)*cos(phi);
> plot3d([X0,Y0,Z0],phi=0..Pi,theta=0..2*Pi,
  scaling=CONSTRAINED, title="Электронное облако");
```



После этого постройте формы электронного облака для $l=1$ и $l=2$.


4. Построить шар $x^2 + y^2 + z^2 = 4$:

```
> with (plots): implicitplot3d(x^2+y^2+z^2=4,  
x=-2..2, y=-2..2, z=-2..2, scaling=CONSTRAINED);
```



5. Построить пространственную кривую: $x = \sin t$, $y = \cos t$, $z = e^t$


```
> with (plots):  
> spacecurve([sin(t), cos(t), exp(t)], t=1..5,  
color=blue, thickness=2, axes=BOXED);
```



6. Нарисовать движущийся объект. Вначале наберите в командной строке:

```
> animate3d(cos(t*x)*sin(t*y), x=-Pi..Pi,
y=-Pi..Pi, t=1..2);
```

Щелкните по появившемуся изображению правой кнопкой мыши. В появившемся контекстном меню выполните команду Animation→Continuous. Затем снова вызовите контекстное меню и выполните команду Animation→Play. Для того, чтобы остановить движение, выполните команду Animation→Stop. Затем с помощью мыши поверните рисунок под другим углом и сделайте его вновь движущимся.



7. Выполните все контрольные задания. Результаты выполнения заданий покажите преподавателю. Сохраните файл со всеми выполненными заданиями на диск. Ответьте на все контрольные вопросы.

Контрольные задания.

1. Построить на отдельных рисунках графики функций Бесселя первого рода $J_n(x)$ для различных ее номеров n в интервале $-20 < x < 20$. Функции Бесселя вызываются командой **BesselJ(n, x)**, где **n** – номер функции Бесселя, **x** – независимая переменная. Построить первые 6 функций Бесселя для $n=0,1,2,3,4,5,6$. Как они выглядят и чем отличаются друг от друга? Сделать подписи осей курсивом.
2. Построить график функции $\rho = \cos^3(\varphi/3)$ в полярных координатах при $0 < \varphi < 4\pi$. Используйте цвет линии под названием **magenta**, установите толщину линии 3.

3. Построить на одном рисунке графики функции $y = x + 2\arcsctg x$ и ее асимптот $y = x$ и $y = x + 2\pi$. Установить следующие параметры: цвет основной линии – голубой, асимптот – красный (установлен по умолчанию, поэтому его можно не изменять); толщина основной линии – 3, асимптоты – обычной; масштаб по координатным осям – одинаковый. Сделать надписи: какая функция относится к какой линии. Указание: использовать для преобразования в текст формул команду **convert**, а для построения графиков и надписей команды **textplot** и **display** из пакета **plots** (см. Задание 1.2, п.2)

4. Нарисовать параметрически заданную поверхность (лист Мебиуса):

$$x = \left(5 + u \cos\left(\frac{v}{2}\right)\right) \cos v, \quad y = \left(5 + u \cos\left(\frac{v}{2}\right)\right) \sin v,$$

$$z = u \sin\left(\frac{v}{2}\right), \quad v \in [0, 2\pi], \quad u \in [-1, 1].$$

5. Задайте изменение координат в интервалах $0 < v < 2\pi$, $-1 < u < 1$, и установите следующие параметры:

grid=[60,10], orientation=[-106,70], axes=FRAMED, tickmarks=[5,8,3].

Также выведите название рисунка, подпишите названия осей и установите одинаковый масштаб по осям.

Контрольные вопросы.

1. С помощью каких команд строятся графики на плоскости и в пространстве? Какие аргументы имеют эти команды?
2. Как называется пакет дополнительных графических команд?
3. С помощью какой команды можно построить график неявной функции? Опишите ее параметры.
4. Для чего предназначена команда **display**?
5. Какая команда позволяет построить двумерную область, заданную системой неравенств?
6. С помощью какой команды можно построить график пространственной кривой?
7. Какие возможности предоставляют команды **animate** и **animate3d**?

IV. Математический анализ: дифференциальное и интегральное исчисление функции одной переменной

1. Вычисление пределов.
2. Дифференцирование.
3. Исследование функции.
4. Интегрирование.

§1. Вычисление пределов

В *Maple* для некоторых математических операций существует по две команды: одна прямого, а другая – отложенного исполнения. Имена команд состоят из одинаковых букв за исключением первой: команды прямого исполнения начинаются со строчной буквы, а команды отложенного исполнения – с заглавной. После обращения к команде отложенного действия математические операции (интеграл, предел, производная и т.д.) выводятся на экран в виде стандартной аналитической записи этой операции. Вычисление в этом случае сразу не производится. Команда прямого исполнения выдает результат сразу.

Для вычисления пределов имеются две команды:

1) прямого исполнения – **limit(expr, x=a, par)**, где **expr** – выражение, предел которого следует найти, **a** – значение точки, для которой вычисляется предел, **par** – необязательный параметр для поиска односторонних пределов (**left** – слева, **right** – справа) или указание типа переменной (**real** – действительная, **complex** – комплексная).

2) отложенного исполнения – **Limit(expr, x=a, par)**, где параметры команды такие же, как и в предыдущем случае. Пример действий этих команд:

```
> Limit(sin(2*x)/x, x=0);  
lim sin(2x)  
x→0 x  
> limit(sin(2*x)/x, x=0);  
2
```

С помощью этих двух команд принято записывать математические выкладки в стандартном аналитическом виде, например:

> **Limit(x*(Pi/2+arctan(x)), x=-infinity) =
limit(x*(Pi/2+arctan(x)), x=-infinity);**

$$\lim_{x \rightarrow \infty} x \left(\frac{\pi}{2} + \arctan(x) \right) = -1$$

Односторонние пределы вычисляются с указанием параметров:
left – для нахождения предела слева и **right** – справа. Например:

> **Limit(1/(1+exp(1/x)), x=0, left) =
limit(1/(1+exp(1/x)), x=0, left);**

$$\lim_{x \rightarrow 0^-} \frac{1}{1 + e^{\frac{1}{x}}} = 1$$

> **Limit(1/(1+exp(1/x)), x=0, right) =
limit(1/(1+exp(1/x)), x=0, right);**

$$\lim_{x \rightarrow 0^+} \frac{1}{1 + e^{\frac{1}{x}}} = 0$$

Задание 1.

1. Вычислить предел $\lim_{x \rightarrow 1} (1-x) \operatorname{tg} \frac{\pi x}{2}$. Наберите:

> **Limit((1-x)*tan(Pi*x/2), x=1) =
limit((1-x)*tan(Pi*x/2), x=1);**

$$\lim_{x \rightarrow 1} (1-x) \tan\left(\frac{1}{2}\pi x\right) = 2 \frac{1}{\pi}$$

2. Найти односторонние пределы $\lim_{x \rightarrow 1^-} \operatorname{arctg} \frac{1}{1-x}$ и $\lim_{x \rightarrow 1^+} \operatorname{arctg} \frac{1}{1-x}$.

Наберите:

> **Limit(arctan(1/(1-x)), x=1, left) =
limit(arctan(1/(1-x)), x=1, left);**

$$\lim_{x \rightarrow 1^-} \arctan\left(\frac{1}{1-x}\right) = \pi \frac{1}{2}$$

> **Limit(arctan(1/(1-x)), x=1, right) =
limit(arctan(1/(1-x)), x=1, right);**

$$\lim_{x \rightarrow 1^+} \arctan\left(\frac{1}{1-x}\right) = -\pi \frac{1}{2}$$

§2. Дифференцирование

Вычисление производных.

Для вычисления производных в *Maple* имеются две команды:

1) прямого исполнения – **diff (f, x)**, где **f** – функция, которую следует продифференцировать, **x** – имя переменной, по которой производится дифференцирование.

2) отложенного исполнения – **Diff (f, x)**, где параметры команды такие же, как и в предыдущей. Действие этой команды сводится к аналитической записи производной в виде $\frac{\partial}{\partial x} f(x)$. После выполнения дифференцирования, полученное выражение желательно упростить. Для этого следует использовать команды **simplify factor** или **expand**, в зависимости от того, в каком виде вам нужен результат.

Пример:

> **Diff (sin (x^2) , x)=diff (sin (x^2) , x) ;**

$$\frac{\partial}{\partial x} \sin(x^2) = 2 \cos(x^2)x$$

Для вычисления производных старших порядков следует указать в параметрах **x\$n**, где **n** – порядок производной; например:

> **Diff (cos (2*x) ^2 , x\$4)=diff (cos (2*x) ^2 , x\$4) ;**

$$\frac{\partial^4}{\partial x^4} \cos(2x)^2 = -128\sin(2x)^2 + 128\cos(2x)^2$$

Полученное выражение можно упростить двумя способами:

> **simplify (%) ;**

$$\frac{\partial^4}{\partial x^4} \cos(2x)^2 = 256\cos(2x)^2 - 128$$

> **combine (%) ;**

$$\frac{\partial^4}{\partial x^4} \left(\frac{1}{2} \cos(4x) + \frac{1}{2} \right)^2 = 128\cos(4x)$$

Дифференциальный оператор.

Для определения дифференциального оператора используется команда **D (f)** – **f**-функция. Например:

> **D (sin) ;**

cos

Вычисление производной в точке:

> **D (sin) (Pi) :eval (%) ;**

-1

Оператор дифференцирования применяется к функциональным операторам

> **f:=x-> ln(x^2)+exp(3*x) :**

> **D (f) ;**

$$x \rightarrow 2\frac{1}{x} + 3e^{(3x)}$$

Задание 2.

1. Вычислить производную $f(x) = \sin^3 2x - \cos^3 2x$

> **Diff (sin (2*x) ^3-cos (2*x) ^3, x) =**

diff (sin (2*x) ^3-cos (2*x) ^3, x) ;

$$\frac{\partial}{\partial x}(\sin(2x)^3 - \cos(2x)^3) = 6\sin(2x)^2 \cos(2x) + 6\cos(2x)^2 \sin(2x)$$

2. Вычислить $\frac{\partial^{24}}{\partial x^{24}}(e^x(x^2 - 1))$. Наберите:

> **Diff (exp (x) * (x^2-1), x\$24) =**

diff (exp (x) * (x^2-1), x\$24) :

> **collect (% , exp (x)) ;**

$$\frac{\partial^{24}}{\partial x^{24}} e^x(x^2 - 1) = e^x(x^2 + 48x + 551)$$

3. Вычислить вторую производную функции $y = \sin^2 x / (2 + \sin x)$ в точках $x=\pi/2, x=\pi$.

> **y:=sin(x)^2/(2+sin(x)) : d2:=diff(y, x\$2) :**

> **x:=Pi ; d2y(x)=d2 ;**

$$x:=\pi \quad d2y(\pi)=1$$

> **x:=Pi/2 ; d2y(x)=d2 ;**

$$x:=\frac{1}{2}\pi \quad d2y\left(\frac{1}{2}\pi\right) = \frac{-5}{9}$$

§3. Исследование функции

Исследование функции необходимо начинать с нахождения ее области определения, но, к сожалению, это трудно автоматизируемая операция. Поэтому при рассмотрении этого вопроса приходится

решать неравенства (см. тему II). Однако, ответить на вопрос, определена ли функция на всей числовой оси, или нет, можно исследовав ее на непрерывность.

Непрерывность функции и точки разрыва.

Проверить непрерывность функции $f(x)$ на заданном промежутке $[x_1, x_2]$ можно с помощью команды **iscont**(**f**, **x=x1..x2**). Если функция **f** непрерывна на этом интервале, то в поле вывода появится ответ *true* – (истина); если функция **f** не является непрерывной на этом интервале, то в поле вывода появится ответ *false* – (ложь). В частности, если задать интервал **x=-infinity..+infinity**, то функция **f** будет проверяться на всей числовой оси. В этом случае, если будет получен ответ *true*, то можно сказать, что функция определена и непрерывна на всей числовой оси. В противном случае следует искать точки разрыва. Это можно сделать двумя способами:

- 1) с помощью команды **discont**(**f**, **x**), где **f** – функция, исследуемая на непрерывность, **x** – переменная. Эта команда пригодна для нахождения точки разрыва первого и второго родов.
- 2) с помощью команды **singular**(**f**, **x**), где **f** – функция, **x** – переменная. Эта команда годится для нахождения точек разрыва второго рода как для вещественных значений переменной, так и для комплексных.

Перед использованием этих команд их следует обязательно загрузить из стандартной библиотеки **readlib**(**name**), где **name** – имя любой из указанных выше команд.

Обе эти команды выдают результаты в виде перечисления точек разрыва в фигурных скобках. Тип такой записи называется *set*. Для того, чтобы в дальнейшем можно было использовать полученные значения точек разрыва, следует из типа *set* с помощью команды **convert** перевести их в обычный числовой тип.

Задание 3.1.

1. Найдите точки разрыва функции $y = e^{\frac{1}{x+3}}$

```
> readlib(iscont): readlib(discont):  
> iscont(exp(1/(x+3)), x=-infinity..+infinity);  
false
```

Это означает, что функция не является непрерывной. Поэтому следует найти точки разрыва с помощью команды:

```
> discont(exp(1/(x+3)), x);
```

{-3}

Ответ наберите в текстовом режиме в новой строке:

“Точка разрыва $x=-3$.”

2. Найдите точки разрыва функции $y = \operatorname{tg} \frac{x}{2-x}$

> **readlib(singular) :**

> **iscont(tan(x/(2-x)), x=-infinity..infinity) ;**

false

> **singular(tan(x/(2-x)), x) ;**

$\{x=2\}, \{x=2 \frac{\pi(2-N+1)}{-2+2-N\pi+\pi}\}$

Здесь $_N$ – целые числа. Ответ наберите в текстовом режиме в новой строке:

“Точки разрыва: $x=2$ и $x=2\pi(2n+1)/(\pi(2n+1)-2)$.”

Экстремумы. Наибольшее и наименьшее значение функции.

В *Maple* для исследования функции на экстремум имеется команда **extrema(f, {cond}, x, 's')**, где **f** – функция, экстремумы которой ищутся, в фигурных скобках **{cond}** указываются ограничения для переменной, **x** – имя переменной, по которой ищется экстремум, в апострофах **'s'** – указывается имя переменной, которой будет присвоена координата точки экстремума. Если оставить пустыми фигурные скобки {}, то поиск экстремумов будет производиться на всей числовой оси. Результат действия этой команды относится к типу *set*. Пример:

> **readlib(extrema) :**

> **extrema(arctan(x) - ln(1+x^2)/2, {}, x, 'x0') ; x0 ;**

$\{\frac{\pi}{4} - \frac{1}{2}\ln(2)\}$

$\{\{x=1\}\}$

В первой строке вывода приводится экстремум функции, а во второй строке вывода – точка этого экстремума.

К сожалению, эта команда не может дать ответ на вопрос, какая из точек экстремума есть максимум, а какая – минимум. Для нахождения максимума функции $f(x)$ по переменной x на интервале $x \in [x1, x2]$ используется команда **maximize(f, x, x=x1..x2)**, а для нахождения минимума функции $f(x)$ по переменной x на интервале $x \in [x1, x2]$ используется команда **minimize(f, x, x=x1..x2)**.

Если после переменной указать **'infinity'** или интервал

$x=-infinity..+infinity$, то команды **maximize** и **minimize** будут искать, соответственно, максимумы и минимумы на всей числовой оси как во множестве вещественных чисел, так и комплексных. Если такие параметры не указывать, то поиск максимумов и минимумов будет производиться только во множестве вещественных чисел. Пример:

```
> maximize (exp (-x^2) , {x}) ;
1
```

Недостаток этих команд в том, что они выдают только значения функции в точках максимума и минимума, соответственно. Поэтому для того, чтобы полностью решить задачу об исследовании функции $y=f(x)$ на экстремумы с указанием их характера (max или min) и координат (x, y) следует сначала выполнить команду:

```
> extrema (f, {}, x, 's') ; s;
```

а затем выполнить команды **maximize (f, x)** ; **minimize (f, x)** . После этого будут полностью найдены координаты всех экстремумов и определены их характеры (max или min).

Команды **maximize** и **minimize** быстро находят абсолютные экстремумы, но не всегда пригодны для нахождения локальных экстремумов. Команда **extrema** вычисляет так же критические точки, в которых функция не имеет экстремума. В этом случае экстремальных значений функции в первой строке вывода будет меньше, чем вычисленных критических точек во второй строке вывода. Выяснить характер найденного экстремума функции $f(x)$ в точке $x=x_0$ можно, если вычислить вторую производную в этой точке и по ее знаку сделать вывод: если $f''(x_0) > 0$, то в точке x_0 будет min, а если $f''(x_0) < 0$ – то max.

В последней версии пакета аналитических вычислений *Maple 6* описанный выше недостаток команд **maximize** и **minimize** устранен. Координаты точек максимума или минимума можно получить, если в параметрах этих команд после переменной записать через запятую новую опцию **location**. В результате в строке вывода после самого максимума (минимума) функции будут в фигурных скобках указаны координаты точек максимума (минимума). Например:

```
> minimize (x^4-x^2, x, location) ;
```

$$\frac{-1}{4}, \left\{ \left[\left\{ x = -\frac{1}{2}\sqrt{2} \right\}, \frac{-1}{4} \right], \left[\left\{ x = \frac{1}{2}\sqrt{2} \right\}, \frac{-1}{4} \right] \right\}$$

В строке вывода получились координаты минимумов и значения функции в этих точках.

Команды **extrema**, **maximize** и **minimize** обязательно должны быть загружены из стандартной библиотеки командой **readlib(name)**, где **name** – имя загружаемой команды.

Задание 3.2.

1. Найти \max и \min $y = \frac{1}{2}(x^2 - \frac{1}{2})\arcsin x + \frac{x}{4}\sqrt{1-x^2} - \frac{\pi}{12}x^2$.

```
> readlib(extrema) :
> y:=(x^2-1/2)*arcsin(x)/2+x*sqrt(1-x^2)/4-
  Pi*x^2/12:
> extrema(y, {}, x, 's'); s;
      {0, -1/24 pi + 1/16 sqrt(3)}
      {{x = 0}, {x = 1/2}}
```

После выполнения этих команд найдены экстремумы функции и точки экстремумов. Порядок следования x -координат экстремумов во второй строке вывода соответствует порядку следования значений экстремумов в первой строке вывода. Таким образом, найдены экстремумы в точках $(0,0)$ и $(1/2, -\pi/24 + \sqrt{3}/16)$. Осталось выяснить, какая из них является максимумом, а какая – минимумом. Для этого используйте команды **maximize** и **minimize**.

```
> readlib(maximize) : readlib(minimize) :
> ymax:=maximize(y, {x});
      ymax := 0
> ymin:=minimize(y, {x});
```

$$ymin := -\frac{1}{24}\pi + \frac{1}{16}\sqrt{3}$$

Ответ наберите в текстовом режиме в новой строке:

“Экстремумы: $\max y(x) = y(0) = 0$,

$\min y(x) = y(1/2) = -\pi/24 + \sqrt{3}/16$.”


Для набора математических символов и греческих букв в текстовом режиме следует нажать кнопку со значком суммы на **Панели инструментов**. В появившейся строке ввода формул ниже **Панели инструментов** следует набирать обычные команды *Maple*, после чего нажать *Enter*. Например, для отображения формулы $\sqrt{3}$ следует набрать в строке ввода формул `sqrt(3)`.

Методы решения математических задач в Maple


Для возвращения в текстовый режим снова следует нажать на кнопку с буквой «Т».



Поэтому порядок набора второй формулы в ответе такой:

- находясь в текстовом режиме, набрать: $\text{miny}(x)=y(1/2)=$;
- нажать на кнопку 

- в строке ввода формул набрать: $-\text{Pi}/24+\text{sqrt}(3)/16$
- нажать *Enter*;



- вернуться в текстовый режим.
2. Найдите наибольшее и наименьшее значение $f(x) = x^2 \ln x$ на интервале $x \in [1, 2]$. Наберите:

```
> f:=x^2*ln(x) :
> maximize(f, {x}, {x=1..2}) ;
      4ln(2)
> minimize(f, {x}, {x=1..2}) : simplify(%);
      -1/2 e^(-1)
```

Ответ наберите в текстовом режиме в новой строке:

”Наибольшее значение: $\max f(x) = 4 \ln 2$, наименьшее значение $\min f(x) = -1/2e$ “.

3. Найти экстремумы функции $y = \frac{x^3}{4-x^2}$ и установить их характер с помощью второй производной. Наберите:

```
> restart:y:=x^3/(4-x^2): readlib(extrema) :
  readlib(maximize) : readlib(minimize) :
> extrema(y, {x}, 's') ; s;
      { -3*sqrt(3), 3*sqrt(3) }
      {{x=0},{x=2*sqrt(3)},{x=-2*sqrt(3)}}
```

Получено два экстремума и три критические точки. Исследование можно продолжить с помощью второй производной:

```
> d2:=diff(y,x$2) : x:=0 : d2y(x) :=d2;
      d2y(0):=0
```


> **x:=2*sqrt(3) : d2y(x) :=d2 ;**

$$d2y(2\sqrt{3}) := -\frac{3}{4}\sqrt{3}$$

> **x:=-2*sqrt(3) : d2y(x) :=d2 ;**

$$d2y(-2\sqrt{3}) := \frac{3}{4}\sqrt{3}$$

Так как $y''(0) = 0$, то в точке $x=0$ нет экстремума; так как $y''(2\sqrt{3}) < 0$, то в точке $x = 2\sqrt{3}$ будет max; так как $y''(-2\sqrt{3}) > 0$, то в точке $x = -2\sqrt{3}$ будет min. Перейдите в текстовый режим и запишите ответ в виде:

“Максимум в точке $(2\sqrt{3}, -3\sqrt{3}/4)$, минимум в точке $(-2\sqrt{3}, 3\sqrt{3}/4)$ ”.

Исследование функции по общей схеме.

1. Область определения функции $f(x)$ – полностью может быть указана после исследования функции на непрерывность.

2. Непрерывность и точки разрыва функции $f(x)$ исследуются по схеме:

> **iscont(f, x=-infinity..infinity) ;**

> **d1:=discont(f, x) ;**

> **d2:=singular(f, x) ;**

В результате наборам переменным **d1** и **d2** будут присвоены значения x -координат в точках разрыва 1 и 2-го родов (если они будут найдены).

3. Асимптоты. Точки бесконечных разрывов определяют вертикальные асимптоты графика $f(x)$. Уравнение вертикальной асимптоты имеет вид:

> **yr:=d2 ;**

Поведение функции $f(x)$ на бесконечности характеризуется наклонными асимптотами (если они есть). Уравнение наклонной асимптоты $y=kx+b$, где коэффициенты вычисляются по формулам:

$$k = \lim_{x \rightarrow +\infty} \frac{f(x)}{x} \quad \text{и} \quad b = \lim_{x \rightarrow +\infty} (f(x) - kx).$$

Аналогичные формулы для $x \rightarrow -\infty$. Поэтому нахождение наклонных асимптот можно провести по следующей схеме:

> **k1:=limit(f(x)/x, x=+infinity) ;**

> **b1:=limit(f(x)-k1*x, x=+infinity) ;**

```
> k2:=limit(f(x)/x, x=-infinity);  
> b2:=limit(f(x)-k2*x, x=-infinity);
```

Часто оказывается, что $k1=k2$ и $b1=b2$, в этом случае будет одна асимптота при $x \rightarrow +\infty$ и при $x \rightarrow -\infty$. С учетом этого составляется уравнение асимптоты

```
> yn:=k1*x+b1;
```

4. Экстремумы. Исследование функции $f(x)$ на экстремумы можно проводить по схеме:

```
> extrema(f(x), {}, x, 's');  
> s;  
> fmax:=maximize(f(x), x);  
> fmin:=minimize(f(x), x);
```

После выполнения этих команд будут найдены координаты (x, y) всех максимумов и минимумов функции $f(x)$.

Построение графика.

Построение графика функции $f(x)$ – это окончательный этап исследования функции. На рисунке помимо графика исследуемой функции $f(x)$ должны быть нанесены все ее асимптоты пунктирными линиями, подписаны координаты точек \max и \min . Приемы построения графиков нескольких функций и нанесения надписей были рассмотрены в теме III.

Задание 3.3.

1. Провести полное исследование функции $f(x) = \frac{x^4}{(1+x)^3}$ по общей схеме. Сначала перейдите в текстовый режим и наберите “Исследование функции: “. Затем вернитесь в режим командной строки и наберите команды:

```
> f:=x^4/(1+x)^3:
```

В текстовом режиме наберите “Непрерывность функции”. В режиме командной строки и наберите:

```
> readlib(iscont): readlib(discont):  
  readlib(singular):  
> iscont(f, x=-infinity..infinity);
```

false

Это означает, что функция не является непрерывной. Перейдите в текстовый режим и наберите “Нахождение точек разрыва”. Вернитесь в режим командной строки и наберите:

```
> discont(f, x);
```

```
{-1}
```

Конвертировать полученное значение точки разрыва типа *set* в число можно командой **convert**, добавив вторую опцию, например, `'+'``. Обратите внимание на обратные кавычки, которые набираются клавишей, расположенной выше клавиши табуляции.

```
> xr:=convert(%, `+'`);
```

```
xr:=-1
```

Перейдите в текстовый режим и наберите: “Получена точка бесконечного разрыва $x=-1$ ”. С новой строки наберите: “Нахождение асимптот.”. Перейдите на новую строку и наберите “Уравнение вертикальной асимптоты: $x=-1$ ” (это можно сделать, поскольку вертикальные асимптоты возникают в точках бесконечного разрыва). С новой строки наберите: “Коэффициенты наклонной асимптоты.”. Перейдите в режим командной строки и наберите:

```
> k1:=limit(f/x, x=infinity);
```

```
k1:=1
```

```
> b1:=limit(f-k1*x, x=infinity);
```

```
b1:=-3
```

```
> k2:=limit(f/x, x=-infinity);
```

```
k2:=1
```

```
> b2:=limit(f-k2*x, x=-infinity);
```

```
b2:=-3
```

В этом случае коэффициенты наклонных асимптот при $x \rightarrow +\infty$ и $x \rightarrow -\infty$ оказались одинаковыми. Поэтому перейдите в текстовый режим и наберите “Уравнение наклонной асимптоты.”. Затем в новой строке перейдите в режим командной строки и наберите:

```
> y=k1*x+b1;
```

```
y = x - 3
```

В текстовом режиме наберите “Нахождение экстремумов”. В новой строке наберите команды:

```
> readlib(extrema): readlib(maximize):
```

```
readlib(minimize):
```

```
> extrema(f, {}, x, 's');s;
```

```
{-256  
 { 27, 0}
```

```
{{x=-4},{x=0}}
```

Поскольку функция имеет разрыв, то при поиске максимума и минимума следует указать интервал, в который не должна входить точка разрыва.

```
> fmax:=maximize(f, {x}, {x=-infinity..-2});
```

$$fmax := \frac{-256}{27}$$

```
> fmin:=minimize(f, {x}, {x=-1/2..infinity});
```

$$fmin := 0$$

В текстовом режиме наберите результат исследования в виде:
 “Максимум в точке $(-4, -256/27)$; минимум в точке $(0, 0)$.”

2. Построить график функции $y = \arctg(x^2)$ и ее асимптоту, указать координаты точек экстремума. Оформление каждого этапа исследования функции проделать также как и при выполнении предыдущего задания. Самостоятельно загрузите из стандартной библиотеки все необходимые команды.

```
> restart: y:=arctan(x^2):
```

```
> iscont(y, x=-infinity..infinity);
```

true

```
> k1:=limit(y/x, x=-infinity);
```

$$k1:=0$$

```
> k2:=limit(y/x, x=+infinity);
```

$$k2:=0$$

```
> b1:=limit(y-k1*x, x=-infinity);
```

$$b1 := \frac{1}{2}\pi$$

```
> b2:=limit(y-k1*x, x=+infinity);
```

$$b2 := \frac{1}{2}\pi$$

```
> yh:=b1;
```

$$yh := \frac{1}{2}\pi$$

```
> extrema(y, {}, x, 's'); s;
```

{0}
{x=0}

```
> ymax:=maximize(y, {x}); ymin:=minimize(y, {x});
```

$$ymax :=$$


$$ymin := 0$$

```
> with(plots): yy:=convert(y, string):
```

```
> p1:=plot(y, x=-5..5, linestyle=1, thickness=3, color=BLACK):
```

```
> p2:=plot(yh, x=-5..5, linestyle=1, thickness=1):
```

```
> t1:=textplot([0.2,1.7,"Асимптота:"],  
font=[TIMES, BOLD, 10], align=RIGHT):  
> t2:=textplot([3.1,1.7,"y=Pi/2"],  
font=[TIMES, ITALIC, 10], align=RIGHT):  
> t3:=textplot([0.1,-0.2,"min:(0,0)"],  
align=RIGHT):  
> t4:=textplot([2,1,yy], font=[TIMES, ITALIC,  
10], align=RIGHT):  
> display([p1,p2,t1,t2,t3,t4]);
```



§4. Интегрирование

Аналитическое и численное интегрирование.

Неопределенный интеграл $\int f(x)dx$ вычисляется с помощью 2-х

команд:

- 1) прямого исполнения – **int(f, x)**, где **f** – подынтегральная функция, **x** – переменная интегрирования;
- 2) отложенного исполнения – **Int(f, x)** – где параметры команды такие же, как и в команде прямого исполнения **int**. Команда **Int** выдает на экран интеграл в аналитическом виде математической формулы.

Для вычисления определенного интеграла $\int_a^b f(x)dx$ в командах

`int` и `Int` добавляются пределы интегрирования, например,

```
> Int((1+cos(x))^2, x=0..Pi)=
int((1+cos(x))^2, x=0..Pi);
```

$$\int_0^{\pi} (1 + \cos(x))^2 dx = \frac{3}{2}\pi$$

Если в команде интегрирования добавить опцию **continuous**: `int(f, x, continuous)`, то *Maple* будет игнорировать любые возможные разрывы подынтегральной функции в диапазоне интегрирования. Это позволяет вычислять несобственные интегралы от неограниченных функций. Несобственные интегралы с бесконечными пределами интегрирования вычисляются, если в параметрах команды `int` указывать, например, `x=0..+infinity`.

Численное интегрирование выполняется командой `evalf(int(f, x=x1..x2), e)`, где **e** – точность вычислений (число знаков после запятой).

Интегралы, зависящие от параметра. Ограничения для параметров.

Если требуется вычислить интеграл, зависящий от параметра, то его значение может зависеть от знака этого параметра или каких-либо других ограничений. Рассмотрим в качестве примера интеграл

$$\int_0^{+\infty} e^{-ax} dx, \text{ который, как известно из математического анализа,}$$

сходится при $a > 0$ и расходится при $a < 0$. Если вычислить его сразу, то получится:

```
> Int(exp(-a*x), x=0..+infinity)=
int(exp(-a*x), x=0..+infinity);
```

Definite integration: Can't determine if the integral is convergent.

Need to know the sign of --> a

Will now try indefinite integration and then take limits.

$$\int_0^{+\infty} e^{(-ax)} dx = \lim_{x \rightarrow \infty} -\frac{e^{(-ax)} - 1}{a}.$$

Таким способом интеграл с параметром не вычислить. Для получения явного аналитического результата вычислений следует сделать какие-либо предположения о значении параметров, то есть наложить на них ограничения. Это можно сделать при помощи команды **assume (expr1)**, где **expr1** – неравенство. Дополнительные ограничения вводятся с помощью команды **additionally (expr2)**, где **expr2** – другое неравенство, ограничивающее значение параметра с другой стороны.

После наложения ограничений на параметр *Maple* добавляет к его имени символ (~), например параметр **a**, на который были наложены некоторые ограничения, в сроке вывода будет иметь вид: $a\sim$.

Описание наложенных ограничений параметра **a** можно вызвать командой **about (a)**. Пример: наложить ограничения на параметр a такие, что $a > -1$, $a \leq 3$:

```
> assume (a>-1) ; additionally (a<=3) ;
> about (a) ;
Originally a, renamed a~:
  is assumed to be: RealRange(Open(-1),3)
```

Вернемся к вычислению интеграла с параметром $\int_0^{+\infty} e^{-ax} dx$,

которое следует производить в таком порядке:

```
> assume (a>0) ;
> Int (exp (-a*x) , x=0..+infinity) =
  int (exp (-a*x) , x=0..+infinity) ;
```

$$\int_0^{+\infty} e^{(-a\sim x)} dx = \frac{1}{a\sim}$$

Обучение основным методам интегрирования.

В *Maple* имеется пакет **student**, предназначенный для обучения математике. Он содержит набор подпрограмм, предназначенных для выполнения расчетов шаг за шагом, так, чтобы была понятна последовательность действий, приводящих к результату. К таким командам относятся интегрирование по частям **inparts** и замена переменной **changevar**.

Формула интегрирования по частям:

$$\int u(x)v'(x)dx = u(x)v(x) - \int u'(x)v(x)dx$$

Методы решения математических задач в Maple

Если обозначить подынтегральную функцию $f=u(x)v'(x)$, то параметры команды интегрирования по частям такие: **intparts** (**Int** (**f**, **x**), **u**), где **u** – именно та функция **u(x)**, производную от которой предстоит вычислить по формуле интегрирования по частям.

Если в интеграле требуется сделать замену переменных $x=g(t)$ или $t=h(x)$, то параметры команды замены переменных такие: **changevar** (**h** (**x**)=**t**, **Int** (**f**, **x**), **t**), где **t** – новая переменная.

Обе команды **intparts** и **changevar** не вычисляют окончательно интеграл, а лишь производят промежуточную выкладку. Для того, чтобы получить окончательный ответ, следует, после выполнения этих команд ввести команду **value** (%); где % - обозначают предыдущую строку.

Не забудьте, перед использованием описанных здесь команд обязательно загрузить пакет **student** командой **with** (**student**) .

Задание 4.

1. Найти неопределенные интегралы: а) $\int \cos x \cos 2x \cos 3x dx$;

б) $\int \frac{3x^4 + 4}{x^2(x^2 + 1)^3} dx$.

> **Int** (**cos** (**x**) * **cos** (**2*x**) * **cos** (**3*x**) , **x**) =
int (**cos** (**x**) * **cos** (**2*x**) * **cos** (**3*x**) , **x**) ;

$$\int \cos(x)\cos(2x)\cos(3x)dx = \frac{1}{8}\sin(2x) + \frac{1}{16}\sin(4x) + \frac{1}{24}\sin(6x) + \frac{1}{4}x$$

> **Int** ((**3*x^4+4**) / (**x^2*(x^2+1)^3**) , **x**) =
int ((**3*x^4+4**) / (**x^2*(x^2+1)^3**) , **x**) ;

$$\int \frac{3x^4 + 4}{x^2(x^2 + 1)^3} dx = -4 \frac{1}{x} - \frac{57}{8} \arctan(x) - \frac{25}{8} \frac{x}{x^2 + 1} - \frac{7}{4} \frac{x}{(x^2 + 1)^2}$$

2. Найти определенный интеграл $\int_0^{\pi/2} \frac{\sin x \cos x dx}{(a^2 \cos^2 x + b^2 \sin^2 x)^2}$, при

условии $a>0, b>0$.

> **assume** (**a>0**) ; **assume** (**b>0**) ;

> **Int** (**sin** (**x**) * **cos** (**x**) / (**a^2*cos** (**x**) ^2 + **b^2*sin** (**x**) ^2) ,
x=0..Pi/2) = **int** (**sin** (**x**) * **cos** (**x**) / (**a^2*cos** (**x**) ^2 + **b^2**

*

$\sin(x)^2, x=0..Pi/2$;

$$\int_0^{\pi/2} \frac{\sin(x)\cos(x)}{(a^2 \cos(x)^2 + b^2 \sin(x)^2)^2} dx = \frac{\ln(b) - \ln(a)}{-a^2 + b^2}$$

3. Найти несобственный интеграл $\int_0^{+\infty} \frac{1 - e^{-ax^2}}{xe^{x^2}} dx$, при $a > -1$

> restart; assume(a>-1);

> Int((1-exp(-a*x^2))/(x*exp(x^2)), x=0..+infinity)=int((1-exp(-a*x^2))/(x*exp(x^2)), x=0..+infinity);

$$\int_0^{+\infty} \frac{1 - e^{-ax^2}}{xe^{x^2}} dx = \frac{1}{2} \ln(a + 1)$$

4. Численно найти интеграл $\int_{\pi/6}^{\pi/4} \frac{\cos x}{x} dx$

> Int(cos(x)/x, x=Pi/6..Pi/4)=evalf(int(cos(x)/x, x=Pi/6..Pi/4), 15);

$$\int_{\pi/6}^{\pi/4} \frac{\cos(x)}{x} dx = .322922981113732$$

5. Полностью проделать все этапы вычисления интеграла

$$\int x^3 \sin x dx \text{ по частям.}$$

> restart; with(student): J:=Int(x^3*sin(x), x);

$$J = \int x^3 \sin(x) dx$$

> J:=intparts(Int(x^3*sin(x), x), x^3);

$$J = -x^3 \cos(x) - \int -3x^2 \cos(x) dx$$

> intparts(%, x^2);

$$J = -x^3 \cos(x) + 3x^2 \sin(x) + \int -6x \sin(x) dx$$

> intparts(%, x);

$$J = -x^3 \cos(x) + 3x^2 \sin(x) + 6x \cos(x) - \int 6 \cos(x) dx$$

> value(%) ;

$$J = -x^3 \cos(x) + 3x^2 \sin(x) + 6x \cos(x) - 6 \sin(x)$$

6. Вычислить интеграл $\int_{-\pi/2}^{\pi/2} \frac{dx}{1 + \cos x}$ с помощью универсальной

подстановки $\operatorname{tg} \frac{x}{2} = t$.

> **J=Int (1/ (1+cos (x)) , x=-Pi/2 .. Pi/2) ;**

$$J = \int_{-\pi/2}^{\pi/2} \frac{1}{1 + \cos(x)} dx$$

> **J=changevar (tan (x/2)=t, Int (1/ (1+cos (x)) , x=-Pi/2 .. Pi/2) , t) ;**

$$J = \int_{-1}^1 \frac{1}{(1 + \cos(2 \arctan(t)))(1 + t^2)} dt$$

> **value (%) ;**

$$J=2$$

Контрольные задания.

1. Вычислите предел $\lim_{x \rightarrow \infty} \left(\frac{x^2 - 2x + 1}{x^2 - 4x + 2} \right)^x$.
2. Найти пределы функции $y = \frac{1}{1 + 2^{1/x}}$ при $x \rightarrow +0$ и при $x \rightarrow -0$.
3. Найти $\frac{\partial^5}{\partial x^5} (\ln x)$.
4. Найти точки разрыва функции $y = \frac{1}{1 - e^{1-x}}$.
5. Найти экстремумы функции $f(x) = x \sin x + \cos x - x^2 / 4$, $x \in [-1, 1]$ и указать их характер.
6. Провести полное исследование функции $y = \frac{x^2(x-1)}{x+1}$.
7. Построить график функции $y = x^3 - 3x^2 + 2$ с указанием координат экстремумов.

8. Вычислить неопределенный интеграл $\int \frac{(x^3 - 6)dx}{x^4 + 6x^2 + 8}$.
9. Вычислить несобственный интеграл $\int_0^{+\infty} \frac{\sin(ax)\cos(bx)dx}{x}$ при $a > 0$
 $b > 0$ для случаев $a > b$, $a = b$, $a < b$.
10. Численно найти интеграл $\int_{0,1}^{0,2} \frac{\sin(3x)e^{-x^2}}{x^4} dx$.
11. Полностью проделать все этапы вычисления интеграла $\int_0^{\pi/2} x^3 \cos x dx$ по частям.
12. Вычислить интеграл $\int_0^{\pi/2} \frac{dx}{5 - 4 \sin x + 3 \cos x}$ с помощью универсальной подстановки $\text{tg}(x/2) = t$.

Контрольные вопросы.

1. Что такое команды прямого и отложенного исполнения? Опишите их действия.
2. С помощью какой команды вычисляются пределы? Какие у нее параметры?
3. Какие команды позволяют найти производную функции?
4. Опишите команды, позволяющие исследовать функцию на непрерывность.
5. Какая последовательность команд необходима для нахождения \max и \min функции с указанием их координат (x, y) ?
6. Какие недостатки имеют команды **maximize**, **minimize** и **extrema**?
7. Опишите общую схему исследования функции и построение ее графика в Maple.
8. Какие команды производят аналитическое и численное интегрирование? Опишите их параметры.
9. С помощью каких команд вводятся ограничения на параметры для вычисления интегралов, зависящих от параметров?
10. Для чего предназначен пакет **student**?
11. Опишите команду интегрирования по частям.
12. Опишите команду интегрирования методом замены переменных.

V. Линейная алгебра

1. Векторная алгебра.
2. Действия с матрицами.
3. Спектральный анализ матрицы.
4. Системы линейных уравнений. Матричные уравнения.

§1. Векторная алгебра

Основная часть команд для решения задач линейной алгебры содержится в библиотеке **linalg**. Поэтому перед решением задач с матрицами и векторами следует загрузить эту библиотеку командой **with(linalg)**.

Способы задания векторов.

Для определения вектора в *Maple* используется команда **vector([x1, x2, ..., xn])**, где в квадратных скобках через запятую указываются координаты вектора. Например:

```
> x:=vector([1, 0, 0]);  
x:=[1, 0, 0]
```

Координату уже определенного вектора **x** можно получить в строке вывода, если ввести команду **x[i]**, где **i** – номер координаты. Например, первую координату заданного в предыдущем примере вектора можно вывести так:

```
> x[1];  
1
```

Вектор можно преобразовать в список и, наоборот, с помощью команды **convert(vector, list)** или **convert(list, vector)**.

Сложение векторов.

Сложить два вектора **a** и **b** можно с помощью двух команд:

- 1) **evalm(a+b)**;
- 2) **matadd(a, b)**.

Команда **add** позволяет вычислять линейную комбинацию векторов **a** и **b**: $\alpha a + \beta b$, где α, β – скалярные величины, если использовать формат: **matadd(a, b, alpha, beta)**.

Скалярное, векторное произведение векторов и угол между векторами.

Скалярное произведение двух векторов $(\mathbf{a}, \mathbf{b}) = \sum_{i=1}^n a_i b_i$

вычисляется командой **dotprod (a , b)** .

Векторное произведение двух векторов $[\mathbf{a}, \mathbf{b}]$ вычисляется командой **crossprod (a , b)** .

Угол между двумя векторами \mathbf{a} и \mathbf{b} вычисляется с помощью команды **angle (a , b)** .

Норма вектора.

Норму (длину) вектора $\mathbf{a} = (x_1, \dots, x_n)$, которая равна $\|\mathbf{a}\| = \sqrt{x_1^2 + \dots + x_n^2}$, можно вычислить с помощью команды **norm (a , 2)** .

Можно нормировать вектор \mathbf{a} с помощью команды **normalize (a)**, в результате выполнения которой будет получен вектор единичной длины $\frac{\mathbf{a}}{\|\mathbf{a}\|}$.

Нахождение базиса системы векторов. Ортогонализация системы векторов по процедуре Грамма-Шмидта.

Если имеется система n векторов $\{\mathbf{a}_1, \mathbf{a}_2, \dots, \mathbf{a}_n\}$, то с помощью команды **basis ([a1 , a2 , ... , an])** можно найти базис этой системы.

При помощи команды **GramSchmidt ([a1 , a2 , ... , an])** можно ортогонализировать систему линейно-независимых векторов $\{\mathbf{a}_1, \mathbf{a}_2, \dots, \mathbf{a}_n\}$.

Задание 1.

1. Даны два вектора: $\mathbf{a} = (2,1,3,2)$ и $\mathbf{b} = (1,2,-2,1)$. Найти (\mathbf{a}, \mathbf{b}) и угол между \mathbf{a} и \mathbf{b} . Для решения этой задачи наберите:

> **with (linalg) :**

> **a := ([2 , 1 , 3 , 2]) ; b := ([1 , 2 , -2 , 1]) ;**

$a := [2, 1, 3, 2]$

$b := [1, 2, -2, 1]$

> **dotprod (a , b) ;**

0

> **phi = angle (a , b) ;**

$$\phi = \frac{\pi}{2}$$

2. Найти векторное произведение $c = [a, b]$, а затем скалярное произведение (a, c) , где $a = (2, -2, 1)$, $b = (2, 3, 6)$.

```
> restart; with(linalg):
> a:=[2,-2,1]; b:=[2,3,6];
      a:=[2,-2,1]
      b:=[2,3,6]
> c:=crossprod(a,b);
      c:=[-15,-10,10]
> dotprod(a,c);
      0
```

3. Найти норму вектора $a = (2, -2, 1)$.

```
> restart; with(linalg):
> a:=vector([1,2,3,4,5,6]): norm(a,2);
      sqrt(91)
```

4. Из системы векторов: $a_1 = (1, 2, 2, -1)$, $a_2 = (1, 1, -5, 3)$, $a_3 = (3, 2, 8, 7)$, $a_4 = (0, 1, 7, -4)$, $a_5 = (2, 1, 12, -10)$ выделить базис и ортогонализировать его по процедуре Грамма-Шмидта:

```
> restart; with(linalg):
> a1:=vector([1,2,2,-1]):
  a2:=vector([1,1,-5,3]):
  a3:=vector([3,2,8,7]): a4:=vector([0,1,7,-4]):
  a5:=vector([2,1,12,-10]):
> g:=basis([a1,a2,a3,a4,a5]);
      g:=[a1, a2, a3, a5]
> GramSchmidt(g);
```

$$\left[[1, 2, 2, -1], [2, 3, -3, 2], \left[\frac{81}{65}, \frac{-93}{65}, \frac{327}{65}, \frac{549}{65} \right], \left[\frac{1633}{724}, \frac{-923}{724}, \frac{-71}{724}, \frac{-355}{724} \right] \right],$$

§2. Действия с матрицами

Определение матрицы.

Для определения матрицы в *Maple* можно использовать команду `matrix(n, m, [[a11,a12,...,a1n], [a21,a22,...,a2m], ...,`

`[an1, an2, ..., anm]]`), где n – число строк, m – число столбцов в матрице. Эти числа задавать необязательно, а достаточно перечислить элементы матрицы построчно в квадратных скобках через запятую. Например:

`> A:=matrix([[1,2,3],[-3,-2,-1]]);`

$$A := \begin{bmatrix} 1 & 2 & 3 \\ -3 & -2 & -1 \end{bmatrix}$$

В Maple матрицы специального вида можно генерировать с помощью дополнительных команд. В частности диагональную матрицу можно получить командой **diag**. Например:

`> J:=diag(1,2,3);`

$$J := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 3 \end{bmatrix}$$

Генерировать матрицу можно с помощью функции $f(i, j)$ от переменных i, j – индексов матрицы: **matrix(n, m, f)**, где n – число строк, m – число столбцов. Например:

`> f:=(i, j)->x^i*y^j;`

$$f := (i, j) \rightarrow x^i y^j$$

`> A:=matrix(2,3,f);`

$$A := \begin{bmatrix} xy & xy^2 & xy^3 \\ x^2y & x^2y^2 & x^2y^3 \end{bmatrix}$$

Число строк в матрице A можно определить с помощью команды **rowdim(A)**, а число столбцов – с помощью команды **coldim(A)**.

Арифметические операции с матрицами.

Сложение двух матриц одинаковой размерности осуществляется теми же командами, что и сложение векторов: **evalm(A+B)** или **matadd(A,B)**. Произведение двух матриц может быть найдено с помощью двух команд:

1) **evalm(A&*B)**;

2) **multiply(A,B)**.

В качестве второго аргумента в командах, вычисляющих произведение, можно указывать вектор, например:

`> A:=matrix([[1,0],[0,-1]]);`

`> B:=matrix([[-5,1],[7,4]]);`

$$A := \begin{bmatrix} 1 & 0 \\ 0 & -1 \end{bmatrix} \quad B := \begin{bmatrix} -5 & 1 \\ 7 & 4 \end{bmatrix}$$

```

> v:=vector([2,4]);
v := [2,4]
> multiply(A,v);
[2,-4]
> multiply(A,B);
[ -5  1
 -7 -4]
> matadd(A,B);
[ -4  1
  7  3]

```

Команда **evalm** позволяет также прибавлять к матрице число и умножать матрицу на число. Например:

```

> C:=matrix([[1,1],[2,3]]);
> evalm(2+3*C);
[ 5  3
  6 11]

```

Определители, миноры и алгебраические дополнения. Ранг и след матрицы.

Определитель матрицы A вычисляется командой **det(A)**. Команда **minor(A,i,j)** возвращает матрицу, полученную из исходной матрицы A вычеркиванием i -ой строки и j -ого столбца. Минор M_{ij} элемента a_{ij} матрицы A можно вычислить командой **det(minor(A,i,j))**. Ранг матрицы A вычисляется командой **rank(A)**. След матрицы A , равный сумме ее диагональных элементов, вычисляется командой **trace(A)**. 7

```

> A:=matrix([[4,0,5],[0,1,-6],[3,0,4]]);
A := [ 4  0  5
       0  1 -6
       3  0  4]
> det(A);
1
> minor(A,3,2);

```



```

                                
$$\begin{bmatrix} 4 & 5 \\ 0 & -6 \end{bmatrix}$$

> det(%);
                                -24
> trace(A);
                                9
```

Обратная и транспонированная матрицы.

Обратную матрицу A^{-1} , такую что $A^{-1}A=AA^{-1}=E$, где E – единичная матрица, можно вычислить двумя способами:

- 1) **evalm**(**1/A**) ;
- 2) **inverse**(**A**) .

Транспонирование матрицы A – это изменение местами строк и столбцов. Полученная в результате этого матрица называется транспонированной и обозначается A' . Транспонированную матрицу A' можно вычислить командой **transpose**(**A**) .

Например, используя заданную в предыдущем пункте матрицу A , найдем ей обратную и транспонированную:

```

> inverse(A) ;
                                
$$\begin{bmatrix} 4 & 0 & -5 \\ -18 & 1 & 24 \\ -3 & 0 & 4 \end{bmatrix}$$

> multiply(A,%);
                                
$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

> transpose(A) ;
                                
$$\begin{bmatrix} 4 & 0 & 3 \\ 0 & 1 & 0 \\ 5 & -6 & 4 \end{bmatrix}$$

```

Выяснение типа матрицы.

Выяснить положительную или отрицательную определенность матрицы можно при помощи команды **definite**(**A,param**) , где **param** может принимать значения: **'positive_def'** – положительно определена ($A>0$), **'positive_semidef'** – неотрицательно определенная ($A \geq 0$), **'negative_def'** –

отрицательно определенная ($A < 0$), **'negative_semidef'** – неположительно определенная ($A \leq 0$). Результатом действия будет константа **true** – подтверждение, **false** – отрицание сделанного предположения. Например:

```
> A:=matrix([[2,1],[1,3]]);
```

$$A := \begin{bmatrix} 2 & 1 \\ 1 & 3 \end{bmatrix}$$

```
> definite(A, 'positive_def');
```

true

Проверить ортогональность матрицы A можно командой **orthog(A)**.

```
> B:=matrix([[1/2,1*sqrt(3)/2],
[1*sqrt(3)/2,-1/2]]);
```

$$B := \begin{bmatrix} \frac{1}{2} & \frac{1}{2}\sqrt{3} \\ \frac{1}{2}\sqrt{3} & -\frac{1}{2} \end{bmatrix}$$

```
> orthog(B);
```

true

Функции от матриц.

Возведение матрицы A в степень n производится командой **evalm(A^n)**. Вычисление матричной экспоненты e^A возможно с помощью команды **exponential(A)**. Например:

```
> T:=matrix([[5*a,2*b],[-2*b,5*a]]);
```

$$T := \begin{bmatrix} 5a & 2b \\ -2b & 5a \end{bmatrix}$$

```
> exponential(T);
```

$$\begin{bmatrix} e^{(5a)} \cos(2b) & e^{(5a)} \sin(2b) \\ -e^{(5a)} \sin(2b) & e^{(5a)} \cos(2b) \end{bmatrix}$$

```
> evalm(T^2);
```

$$\begin{bmatrix} 25a^2 - 4b^2 & 20ab \\ -20ab & 25a^2 - 4b^2 \end{bmatrix}$$

Задание 2.

1. Даны матрицы: $A = \begin{bmatrix} 4 & 3 \\ 7 & 5 \end{bmatrix}$, $B = \begin{bmatrix} -28 & 93 \\ 38 & -126 \end{bmatrix}$, $C = \begin{bmatrix} 7 & 3 \\ 2 & 1 \end{bmatrix}$. Найти:

$(AB)C$, $\det A$, $\det B$, $\det C$, $\det[(AB)C]$. Наберите:

```
> with(linalg): restart;
> A:=matrix([[4,3],[7,5]]):
> B:=matrix([[-28,93],[38,-126]]):
> C:=matrix([[7,3],[2,1]]):
> F:=evalm(A&*B&*C);
```

$$F = \begin{bmatrix} 2 & 0 \\ 0 & 3 \end{bmatrix}$$

```
> Det(A)=det(A); Det(B)=det(B); Det(C)=det(C);
  Det(F)=det(F);
```

$$\text{Det}(A)=-1$$

$$\text{Det}(B)=-6$$

$$\text{Det}(C)=1$$

$$\text{Det}(F)=6$$

2. Дана матрица $A = \begin{bmatrix} 2 & 5 & 7 \\ 6 & 3 & 4 \\ 5 & -2 & -3 \end{bmatrix}$, найти: $\det A$, A^{-1} , A' , $\det(M_{22})$.

Наберите:

```
> A:=matrix([[2,5,7],[6,3,4],[5,-2,-3]]);
```

$$A := \begin{bmatrix} 2 & 5 & 7 \\ 6 & 3 & 4 \\ 5 & -2 & -3 \end{bmatrix}$$

```
> Det(A)=det(A);
```

$$\text{Det}(A)=-1$$

```
> transpose(A);
```

$$\begin{bmatrix} 2 & 6 & 5 \\ 5 & 3 & -2 \\ 7 & 4 & -3 \end{bmatrix}$$

```
> inverse(A);
```

$$\begin{bmatrix} 1 & -1 & 1 \\ -38 & 41 & -34 \\ 27 & -29 & 24 \end{bmatrix}$$

> **det (minor (A, 2, 2)) ;**

-41

3. Найти ранг матрицы $A = \begin{bmatrix} 8 & -4 & 5 & 5 & 9 \\ 1 & -3 & -5 & 0 & -7 \\ 7 & -5 & 1 & 4 & 1 \\ 3 & -1 & 3 & 2 & 5 \end{bmatrix}$.

> **A:=matrix ([[8,-4,5,5,9], [1,-3,-5,0,-7], [7,-5,1,4,1], [3,-1,3,2,5]]);**

> **r (A)=rank (A) ;**

r(A)=3

4. Вычислить e^T , где $T = \begin{bmatrix} 3 & -1 \\ 1 & 1 \end{bmatrix}$.

> **exponential ([[3,-1],[1,1]]);**

$$\begin{bmatrix} 2e^2 & -e^2 \\ e^2 & 0 \end{bmatrix}$$

5. Дана матрица $A = \begin{bmatrix} 5 & 1 & 4 \\ 3 & 3 & 2 \\ 6 & 2 & 10 \end{bmatrix}$. Найти значение многочлена

$$P(A) = A^3 - 18A^2 + 64A.$$

> **A:=matrix ([[5,1,4],[3,3,2],[6,2,10]]);**

> **P (A)=evalm (A^3-18*A^2+64*A) ;**

$$P(A) = \begin{bmatrix} 64 & 0 & 0 \\ 0 & 64 & 0 \\ 0 & 0 & 64 \end{bmatrix}$$

§3. Спектральный анализ матрицы

Собственные числа и собственные векторы матрицы.

Из курса линейной алгебры известно, что если $Ax = \lambda x$, то вектор x называется собственным вектором матрицы A , а число λ –

собственным числом, соответствующим данному собственному вектору. Совокупность всех собственных чисел матрицы называется спектром матрицы. Если в спектре матрицы одно и тоже собственное число встречается k раз, то говорят, что кратность этого собственного числа равна k .

Для нахождения собственных чисел матрицы A используется команда **eigenvalues (A)**. Для нахождения собственных векторов матрицы A используется команда **eigenvectors (A)**. В результате выполнения этой команды будут получены собственные числа, их кратность и соответствующие собственные векторы.

Чтобы понять, в каком виде получаются результаты выполнения команды **eigenvectors**, внимательно разберитесь со следующим

примером: матрица $A = \begin{bmatrix} 3 & -1 & 1 \\ -1 & 5 & -1 \\ 1 & -1 & 3 \end{bmatrix}$ имеет 3 собственных вектора:

$a_1 = (-1, 0, 1)$, отвечающий собственному числу $\lambda_1 = 2$ кратности 1,

$a_2 = (1, 1, 1)$, отвечающий собственному числу $\lambda_2 = 3$ кратности 1,

$a_3 = (1, -2, 1)$, отвечающий собственному числу $\lambda_3 = 6$ кратности 1.

Найдем их в Maple:

```
> A:=matrix([[3,-1,1],[-1,5,-1],[1,-1,3]]):
```

```
> eigenvectors(A);
```

```
[2,1,{{-1,0,1}}, [3,1,{{1,1,1}}, [6,1,{{1,-2,1}}]
```

В строке вывода перечислены в квадратных скобках собственное число, его кратность и соответствующий собственный вектор в фигурных скобках, затем следующие наборы таких же данных.

Характеристический и минимальный многочлены матрицы.

Для вычисления характеристического многочлена $P_A(\lambda) = \det(\lambda E - A)$ матрицы A используется команда **charpoly (A, lambda)**.

Минимальный многочлен (делитель) матрицы A можно найти с помощью команды **minpoly (A, lambda)**.

Канонические и специальные виды матрицы.

Привести матрицу A к нормальной форме Жордана можно командой **jordan (A)**.

К треугольному виду матрицу A можно привести тремя способами:

- 1) команда **gausselim(A)** приводит матрицу A к треугольному виду методом Гаусса;
- 2) команда **ffgausselim(A)** приводит матрицу A к треугольному виду методом Гаусса без деления. Эта команда предпочтительней для работы с символьными матрицами, так как не производит нормировку элементов и исключает возможные ошибки, связанные с делением на нуль;
- 3) команда **gaussjord(A)** приводит матрицу A к треугольному виду методом Гаусса-Жордана.

Характеристическую матрицу $F(A) = \lambda E - A$ можно вычислить командой **charmat(A, lambda)**.

Задание 3.

1. Дана матрица $U = \begin{bmatrix} 3 & 2-i \\ 2+i & 7 \end{bmatrix}$. Найти ее собственные векторы и собственные числа.

> **U:=matrix([[3, 2-I], [2+I, 7]]):**

> **eigenvectors(U);**

$$\left[8, 1, \left\{ \frac{2}{5} - \frac{1}{5}I, 1 \right\} \right], [2, 1, \{-2 + I, 1\}]$$

2. Дана матрица $A = \begin{bmatrix} 3 & -i & 0 \\ i & 3 & 0 \\ 0 & 0 & 4 \end{bmatrix}$. Найти собственные векторы,

собственные числа, характеристический многочлен и минимальный многочлен, Жорданову форму.

> **A:=matrix([[3, -I, 0], [I, 3, 0], [0, 0, 4]]):**

> **eigenvectors(A);**

$$[2, 1, \{([1, -I, 0])\}], [4, 2, \{([0, 0, 1]), ([-I, 1, 0])\}]$$

> **P(lambda):=charpoly(A, lambda);**

$$P(\lambda) := \lambda^3 - 10\lambda^2 + 32\lambda - 32$$

> **d(lambda):=minpoly(A, lambda);**

$$d(\lambda) := 8 - 6\lambda + \lambda^2$$

> **jordan(A);**

$$\begin{bmatrix} 2 & 0 & 0 \\ 0 & 4 & 0 \\ 0 & 0 & 4 \end{bmatrix}$$

3. Дана матрица $A = \begin{bmatrix} 1 & -3 & 4 \\ 4 & -7 & 8 \\ 6 & -7 & 7 \end{bmatrix}$. Привести матрицу A к Жордановой

форме, треугольному виду, найти ее характеристическую матрицу.

> **A:=matrix**([[1,-3,4], [4,-7,8], [6,-7,7]]) :

> **j:=jordan**(A) ;

$$j := \begin{bmatrix} 3 & 0 & 0 \\ 0 & -1 & 1 \\ 0 & 0 & -1 \end{bmatrix}$$

> **g:=gausselim**(A) ;

$$g := \begin{bmatrix} 1 & -3 & 4 \\ 0 & 5 & -8 \\ 0 & 0 & \frac{3}{5} \end{bmatrix}$$

> **F(A) := charmat**(A, lambda) ;

$$F(A) := \begin{bmatrix} \lambda - 1 & 3 & -4 \\ -4 & \lambda + 7 & -8 \\ -6 & 7 & \lambda - 7 \end{bmatrix}$$

Самостоятельно проверьте, чем будет отличаться результат выполнения команды **ffgausselim**(A) от **gausselim**(A) на этом примере.

§4. Системы линейных уравнений. Матричные уравнения

Системы линейных уравнений и матричные уравнения.

Система линейных уравнений $Ax = b$ может быть решена двумя способами.

Способ 1: стандартная команда **solve** находит решение системы линейных уравнений, записанных в развернутом виде:

> **X:=linsolve(A,B);**

$$X := \begin{bmatrix} -1 & -1 \\ 2 & 3 \end{bmatrix}$$

3. Дана матрица $A = \begin{bmatrix} 1 & 1 & 0 \\ 0 & 2 & -1 \\ 1 & 3 & -1 \end{bmatrix}$. Найти ее ранг, дефект: $d(A)=n-r(A)$,

где n – размерность квадратной матрицы, r – ее ранг. Найти ядро A . Наберите:

> **A:=matrix([[1,1,0],[0,2,-1],[1,3,-1]]):**

> **r(A):=rank(A);**

$$r(A):=2$$

> **d(A):=rowdim(A)-r(A);**

$$d(A):=1$$

> **k(A):=kernel(A);**

$$k(A):=\{[-1,1,2]\}$$

Контрольные задания.

- 1) Даны 2 вектора: $\mathbf{a} = (1, 2, 2, 3)$, $\mathbf{b} = (3, 1, 5, 1)$. Найти (\mathbf{a}, \mathbf{b}) и угол ϕ между этими векторами.
- 2) Даны 3 вектора: $\mathbf{a} = (2, -3, 1)$, $\mathbf{b} = (-3, 1, 2)$ и $\mathbf{c} = (1, 2, 3)$. Найти: $[[\mathbf{a}, \mathbf{b}], \mathbf{c}]$ и $[\mathbf{a}, [\mathbf{b}, \mathbf{c}]]$.
- 3) Даны системы векторов: $\mathbf{a}_1 = (2, 1, 3, -1)$, $\mathbf{a}_2 = (7, 4, 3, -3)$, $\mathbf{a}_3 = (1, 1, -6, 0)$, $\mathbf{a}_4 = (5, 3, 0, 4)$. Предварительно выяснив, является ли система $\{\mathbf{a}_1, \mathbf{a}_2, \mathbf{a}_3, \mathbf{a}_4\}$ базисом, применить процедуру ортогонализации Грамма-Шмидта и построить ортогональный базис этого подпространства.

- 4) Даны матрицы $A = \begin{bmatrix} 5 & 7 & -3 & -4 \\ 7 & 6 & -4 & -5 \\ 6 & 4 & -3 & -2 \\ 8 & 5 & -6 & -1 \end{bmatrix}$ и $B = \begin{bmatrix} 1 & 2 & 3 & 4 \\ 2 & 3 & 4 & 5 \\ 1 & 3 & 5 & 7 \\ 2 & 4 & 6 & 8 \end{bmatrix}$. Найти:

$AB, BA, \det A, \det B$.

5) Дана матрица: $A = \begin{bmatrix} 1 & 2 & 3 & 4 \\ 2 & 3 & 1 & 2 \\ 1 & 1 & 1 & -1 \\ 1 & 0 & -2 & -6 \end{bmatrix}$. Найти: $\det A$, A^{-1} , M_{32} , A' .

6) Найти ранг матрицы: $C = \begin{bmatrix} -6 & 4 & 8 & -1 & 6 \\ -5 & 2 & 4 & 1 & 3 \\ 7 & 2 & 4 & 1 & 3 \\ 2 & 4 & 8 & -7 & 6 \\ 3 & 2 & 4 & -5 & 3 \end{bmatrix}$. Привести матрицу

C к треугольному виду.

7) Дана матрица $A = \begin{bmatrix} 5 & 4 & 3 & 2 & 1 \\ 4 & 8 & 6 & 4 & 2 \\ 3 & 6 & 9 & 6 & 3 \\ 2 & 4 & 6 & 8 & 4 \\ 1 & 2 & 3 & 4 & 5 \end{bmatrix}$. Найти ее спектр,

характеристический многочлен и значение матрицы на нем (вместо переменной λ в $P_A(\lambda)$ подставить A).

8) Дана матрица $T = \begin{bmatrix} 4 & 2 & -5 \\ 6 & 4 & -9 \\ 5 & 3 & -7 \end{bmatrix}$. Найти e^T , $\det(e^T)$, собственные

векторы и собственные числа матрицы e^T , ядро матрицы T .

9) Дана матрица $U = \begin{bmatrix} 3 & -4 & 0 & 2 \\ 4 & -5 & -2 & 4 \\ 0 & 0 & 3 & -2 \\ 0 & 0 & 2 & -1 \end{bmatrix}$. Найти нормальную форму

Жордана, собственные векторы и числа, найти характеристический и минимальный многочлены.

10) Решить матричное уравнение: $AX=B$, где $A = \begin{bmatrix} 1 & 2 & -3 \\ 3 & 2 & -4 \\ 2 & -1 & 0 \end{bmatrix}$,

$$B = \begin{bmatrix} 1 & -3 & 0 \\ 10 & 2 & 7 \\ 10 & 7 & 8 \end{bmatrix}.$$

Контрольные вопросы.

1. Какой пакет следует загрузить перед решением задач линейной алгебры в *Maple*?
2. С помощью каких команд можно ввести вектор, матрицу?
3. Какими двумя командами можно сложить два вектора одинаковой размерности (2 матрицы)?
4. Какие виды произведений векторов вычисляются *Maple* и какие команды для этого используются?
5. Как вычислить норму вектора?
6. Как вычислить угол между двумя векторами?
7. Опишите команды нахождения базиса системы векторов и построение ортогонального базиса системы векторов.
8. Какими двумя командами можно вычислить произведение двух матриц (или матрицы на вектор)?
9. Какие команды используются для нахождения определителя, минора, алгебраического дополнения, следа матрицы?
10. Что такое дефект матрицы? Опишите способ нахождения дефекта квадратной матрицы. Какие команды при этом используются?
11. Какая матрица называется обратной и какими способами она вычисляется в *Maple*?
12. Что называется собственным вектором и собственным числом матрицы? Что называется спектром матрицы? Какие команды используются для нахождения спектра матрицы и ее собственных векторов? В каком виде в *Maple* выводятся результаты выполнения этих команд?
13. Перечислите специальные виды матриц и команды, приводящие матрицы к этим формам.
14. Что называется ядром матрицы, и какая команда используется для его нахождения?
15. Какая команда позволяет решать матричные уравнения?

VI. Дифференциальные уравнения

1. Аналитическое решение дифференциальных уравнений.
2. Численное решение дифференциальных уравнений.

§1. Аналитическое решение дифференциальных уравнений

Общее решение дифференциальных уравнений.

Для нахождения аналитических решений дифференциальных уравнений в *Maple* применяется команда **dsolve(eq, var, options)**, где **eq** – дифференциальное уравнение, **var** – неизвестные функции, **options** – параметры. Параметры могут указывать метод решения задачи, например, по умолчанию ищется аналитическое решение: **type=exact**. При составлении дифференциальных уравнений для обозначения производной применяется команда **diff**, например, дифференциальное уравнение $y''+y=x$ записывается в виде: **diff(y(x), x\$2)+y(x)=x**.

Общее решение дифференциального уравнения зависит от произвольных постоянных, число которых равно порядку дифференциального уравнения. В *Maple* такие постоянные, как правило, обозначаются как **_C1**, **_C2**, и т.д.

Общее решение неоднородного линейного дифференциального уравнения всегда выводится так, чтобы была четко видна, структура этого решения. Как известно, общее решение неоднородного линейного дифференциального уравнения равно сумме общего решения соответствующего однородного дифференциального уравнения и частного решения этого же неоднородного дифференциального уравнения. Поэтому в строке вывода решение неоднородного линейного дифференциального уравнения всегда состоит из слагаемых, которые содержат произвольные постоянные (это общее решения соответствующего однородного дифференциального уравнения), и слагаемых без произвольных постоянных (это частное решения этого же неоднородного дифференциального уравнения).

Команда **dsolve** выдает решение дифференциального уравнения в невычисляемом формате. Для того, чтобы с решением можно было бы работать далее (например, построить график решения) следует отделить правую часть полученного решения командой **rhs(%)**.

Задание 1.1.

1. Найти общее решение дифференциального уравнения $y' + y \cos x = \sin x \cos x$.

> **restart;**

> **de:=diff(y(x),x)+y(x)*cos(x)=sin(x)*cos(x);**

$$de := \left(\frac{\partial}{\partial x} y(x) \right) + y(x) \cos(x) = \sin(x) \cos(x)$$

> **dsolve(de,y(x));**

$$y(x) = \sin(x) - 1 + e^{(-\sin(x))} _C1$$

Итак, решение искомого уравнения есть функция

$$y(x) = \sin(x) - 1 + e^{(-\sin(x))} _C1.$$

Замечание: при записи решения дифференциального уравнения в Maple в строке вывода произвольная постоянная обозначена как $_C1$.

2. Найти общее решение дифференциального уравнения второго порядка $y'' - 2y' + y = \sin x + e^{-x}$.

> **restart;**

> **deq:=diff(y(x),x\$2)-2*diff(y(x),x)+y(x)=sin(x)+exp(-x);**

$$deq := \left(\frac{\partial^2}{\partial x^2} y(x) \right) - 2 \left(\frac{\partial}{\partial x} y(x) \right) + y(x) = \sin(x) + e^{(-x)}$$

> **dsolve(deq,y(x));**

$$y(x) = _C1 e^x + _C2 e^x x + \frac{1}{2} \cos(x) + \frac{1}{4} e^{(-x)}$$

Замечание: так как исходное уравнение было второго порядка, то полученное решение содержит две произвольные константы, которые в Maple обычно обозначаются как $_C1$ и $_C2$. Первые два слагаемых представляют собой общее решение соответствующего однородного дифференциального уравнения, а вторые два – частное решение неоднородного дифференциального уравнения.

3. Найти общее решение дифференциального уравнения порядка $y'' + k^2 y = \sin(qx)$ в двух случаях: $q \neq k$ и $q = k$ (резонанс).

> **restart; de:=diff(y(x),x\$2)+k^2*y(x)=sin(q*x);**

$$de := \left(\frac{\partial^2}{\partial x^2} y(x) \right) + k^2 y(x) = \sin(qx)$$

> **dsolve(deq,y(x));**

$$y(x) = \frac{\left(-\frac{1}{2} \frac{\cos((k+q)x)}{k+q} + \frac{1}{2} \frac{\cos((k-q)x)}{k-q} \right) \sin(kx) - \left(\frac{1}{2} \frac{\sin((k-q)x)}{k-q} - \frac{1}{2} \frac{\sin((k+q)x)}{k+q} \right) \cos(kx)}{k} + _C1 \sin(kx) + _C2 \cos(kx)$$

Теперь найдем решение в случае резонанса. Для этого перед вызовом команды **dsolve** следует приравнять $q=k$.

> **q:=k: dsolve(de, y(x));**

$$y(x) = -\frac{1}{2} \frac{\cos(kx)^2 \sin(kx)}{k^2} - \frac{\left(-\frac{1}{2} \cos(kx) \sin(kx) + \frac{1}{2} kx \right) \cos(kx)}{k^2} + _C1 \sin(kx) + _C2 \cos(kx)$$

Замечание: в обоих случаях частное решение неоднородного уравнения и общее решение, содержащее произвольные постоянные, выводятся отдельными слагаемыми.

Фундаментальная (базисная) система решений.

Команда **dsolve** представляет возможность найти фундаментальную систему решений (базисные функции) дифференциального уравнения. Для этого в параметрах команды **dsolve** следует указать **output=basis**.

Задание 1.2.

Найти фундаментальную систему решений дифференциального уравнения: $y^{(4)} + 2y'' + y = 0$.

> **de:=diff(y(x), x\$4)+2*diff(y(x), x\$2)+y(x)=0;**

$$de := \left(\frac{\partial^4}{\partial x^4} y(x) \right) + 2 \left(\frac{\partial^2}{\partial x^2} y(x) \right) + y(x) = 0$$

> **dsolve(de, y(x), output=basis);**

$$[\cos(x), \sin(x), x \cos(x), x \sin(x)]$$

Решение задачи Коши или краевой задачи.

Команда **dsolve** может найти решение задачи Коши или краевой задачи, если помимо дифференциального уравнения задать начальные или краевые условия для неизвестной функции. Для обозначения производных в начальных или краевых условиях

используется дифференциальный оператор D , например, условие $y''(0)=2$ следует записать в виде $(D@@2)(y)(0)=2$, или условие $y'(1)=0$: $D(y)(1)=0$. Напомним, что производная n -го порядка записывается в виде $(D@@n)(y)$.

Задание 1.3.

1. Найти решение задачи Коши: $y^{(4)}+y''=2\cos x$, $y(0)=-2$, $y'(0)=1$, $y''(0)=0$, $y'''(0)=0$.

> **de:=diff (y (x) , x\$4)+diff (y (x) , x\$2)=2*cos (x) ;**

$$de := \left(\frac{\partial^4}{\partial x^4} y(x) \right) + \left(\frac{\partial^2}{\partial x^2} y(x) \right) = 2 \cos(x)$$

> **cond:=y (0)=-2, D (y) (0)=1, (D@@2) (y) (0)=0, (D@@3) (y) (0)=0 ;**

$$cond := y(0)=-2, D(y)(0)=1, (D^{(2)})(y)(0)=0, (D^{(3)})(y)(0)=0$$

> **dsolve ({de, cond} , y (x)) ;**

$$y(x)=-2\cos(x)-x\sin(x)+x$$

2. Найти решение краевой задачи: $y''+y = 2x - \pi$, $y(0) = 0$, $y\left(\frac{\pi}{2}\right) = 0$.

Построить график решения.

> **restart ; de:=diff (y (x) , x\$2)+y (x)=2*x-Pi ;**

$$de := \left(\frac{\partial^2}{\partial x^2} y(x) \right) + y(x) = 2x - \pi$$

> **cond:=y (0)=0, y (Pi/2)=0 ;**


$$cond := y(0) = 0, y\left(\frac{\pi}{2}\right) = 0$$

> **dsolve ({de, cond} , y (x)) ;**

$$y(x)=2x-\pi+\pi\cos(x)$$

Замечание: для построения графика решения предварительно следует отделить правую часть полученного выражения.

> **y1:=rhs (%) :plot (y1, x=-10..20, thickness=2) ;**



Системы дифференциальных уравнений.

Команда **dsolve** может найти решение системы дифференциальных уравнений (или задачи Коши), если в ней указать: **dsolve({sys}, {x(t), y(t), ...})**, где **sys** – система дифференциальных уравнений, **x(t), y(t), ...** – набор неизвестных функций.

Задание 1.4.

Найти решение системы дифференциальных уравнений:

$$\begin{cases} x' = -4x - 2y + \frac{2}{e^t - 1}, \\ y' = 6x + 3y - \frac{3}{e^t - 1} \end{cases}$$

```
> sys:=diff(x(t), t)=-4*x(t)-2*y(t)+2/(exp(t)-1),
diff(y(t), t)=6*x(t)+3*y(t)-3/(exp(t)-1):
```

```
> dsolve({sys}, {x(t), y(t)});
```

$$\{x(t) = -3_C1 + 4C1_e^{(-t)} - 2C2_ + 2C2_e^{(-t)} + 2e^{(-t)} \ln(e^t - 1),$$

$$y(t) = 6_C1 - 6_C1e^{-t} - 3_C2e^{(-t)} + 4_C2 - 3e^{(-t)} \ln(e^t - 1)\}$$

Найдены две функции $x(t)$ и $y(t)$, которые зависят от двух произвольных постоянных $_C1$ и $_C2$.

Приближенное решение дифференциальных уравнений с помощью степенных рядов.

Для многих типов дифференциальных уравнений не может быть найдено точное аналитическое решение. В этом случае дифференциальное уравнение можно решить с помощью

приближенных методов, и, в частности, с помощью разложения в степенной ряд неизвестной функции.

Чтобы найти приближенное решение дифференциального уравнения в виде степенного ряда, в команде **dsolve** следует после переменных указать параметр **type=series** (или просто **series**). Для того, чтобы указать порядок разложения **n**, т.е. порядок степени, до которой производить разложение, следует перед командой **dsolve** вставить определение порядка с помощью команды **Order:=n**.

Если ищется общее решение дифференциального уравнения в виде разложения в степенной ряд, то коэффициенты при степенях x найденного разложения будут содержать неизвестные значения функции в нуле $\mathbf{y}(0)$ и ее производных $\mathbf{D}(\mathbf{y})(0)$, $\mathbf{D}@@2(\mathbf{y})(0)$ и т.д. Полученное в строке вывода выражение будет иметь вид, похожий на разложение искомого решения в ряд Маклорена, но с другими коэффициентами при степенях x . Для выделения частного решения следует задать начальные условия $\mathbf{y}(0)=\mathbf{y1}$, $\mathbf{D}(\mathbf{y})(0)=\mathbf{y2}$, $\mathbf{D}@@2(\mathbf{y})(0)=\mathbf{y3}$ и т.д., причем количество этих начальных условий должно совпадать с порядком соответствующего дифференциального уравнения.

Разложение в степенной ряд имеет тип **series**, поэтому для дальнейшей работы с этим рядом его следует преобразовать в полином с помощью команды **convert(% ,polynom)**, а затем выделить правую часть полученного выражения командой **rhs(%)**.

Задание 1.5.

1. Найти решение задачи Коши: $y' = y + xe^y$, $y(0) = 0$ в виде степенного ряда с точностью до 5-го порядка.

```
> restart; Order:=5;
> dsolve({diff(y(x),x)=y(x)+x*exp(y(x)),
  y(0)=0}, y(x), type=series);
```

$$y(x) = \frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{1}{6}x^4 + O(x^5)$$

В полученном решении слагаемое $O(x^5)$ означает, что точность разложения была до 5-го порядка.

2. Найти общее решение дифференциального уравнения $y''(x)-y^3(x)=e^{-x}\cos x$, в виде разложения в степенной ряд до 4-го порядка. Найти разложение при начальных условиях: $y(0)=1$, $y'(0)=0$.

```
> restart; Order:=4: de:=diff(y(x),x$2)-
  y(x)^3=exp(-x)*cos(x):
```

> **f:=dsolve(de, y(x), series);**

$$f := y(x) = y(0) + D(y)(0)x + \left(\frac{1}{2}y(0)^3 + \frac{1}{2}\right)x^2 + \left(\frac{1}{2}y(0)^2 D(y)(0) - \frac{1}{6}\right)x^3 + O(x^4)$$

Замечание: в полученном разложении запись **D(y)(0)** обозначает производную в нуле: $y'(0)$. Для нахождения частного решения осталось задать начальные условия:

> **y(0):=1: D(y)(0):=0:f;**

$$y(x) = 1 + x^2 - \frac{1}{6}x^3 + O(x^4)$$

3. Найти приближенное решение в виде степенного ряда до 6-го порядка и точное решение задачи Коши: $y''' - y' = 3(2 - x^2)\sin x$, $y(0) = 1$, $y'(0) = 1$, $y''(0) = 1$. Построить на одном рисунке графики точного и приближенного решений.

> **restart; Order:=6:**

> **de:=diff(y(x), x\$3)-diff(y(x), x)=3*(2-x^2)*sin(x);**

$$de := \left(\frac{\partial^3}{\partial x^3} y(x)\right) - \left(\frac{\partial}{\partial x} y(x)\right) = 3(2 - x^2)\sin(x)$$

> **cond:=y(0)=1, D(y)(0)=1, (D@@2)(y)(0)=1;**
cond:=y(0)=1, D(y)(0)=1, D⁽²⁾(y)(0)=1

> **dsolve({de, cond}, y(x));**

$$y(x) = \frac{21}{2}\cos(x) - \frac{3}{2}x^2\cos(x) + 6x\sin(x) - 12 + \frac{7}{4}e^x + \frac{3}{4}e^{-x}$$

> **y1:=rhs(%):**

> **dsolve({de, cond}, y(x), series);**

$$y(x) = 1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{7}{24}x^4 + \frac{1}{120}x^5 + O(x^6)$$


Замечание: тип решения дифференциального уравнения в виде ряда есть **series**, поэтому для дальнейшего использования такого решения (вычислений или построения графика) его обязательно следует конвертировать в полином с помощью команды **convert**

> **convert(%, polynomial): y2:=rhs(%):**

> **p1:=plot(y1, x=-3..3, thickness=2, color=black):**

> **p2:=plot(y2, x=-3..3, linestyle=3, thickness=2, color=blue):**

```
> with(plots): display(p1,p2);
```



На этом рисунке видно, что наилучшее приближение точного решения степенным рядом достигается примерно на интервале $-1 < x < 1$.

§2. Численное решение дифференциальных уравнений

Численное решение дифференциальных уравнений с помощью команды dsolve. Построение графиков решений дифференциальных уравнений с помощью команды odeplot.

Для того, чтобы найти численное решение дифференциального уравнения (задачи Коши или краевой задачи) в команде **dsolve** следует указать параметр **type=numeric** (или просто **numeric**). Тогда команда решения дифференциального уравнения будет иметь вид **dsolve(eq, vars, type=numeric, options)**, где **eq** – уравнения, **vars** – список неизвестных функций, **options** – параметры, позволяющие указать метод численного интегрирования дифференциального уравнения. В *Maple* реализованы такие методы: **method=rkf45** – метод Рунге-Кутты-Фельберга 4-5-ого порядка (установлен по умолчанию); **method=dverk78** – метод Рунге-Кутты 7-8 порядка; **method=classical** – классический метод Рунге-Кутты 3-его порядка; **method=gear** и **method=mgear** – одношаговый и многошаговый методы Гира.

График численного решения дифференциального уравнения можно построить с помощью команды **odeplot(dd, [x, y(x)], x=x1..x2)**, где в качестве функции используется команда **dd:=dsolve({eq, cond}, y(x), numeric)** численного решения, после нее в квадратных скобках указывают переменную и

неизвестную функцию $[x, y(x)]$, и интервал $x=x1..x2$ для построения графика.

Задание 2.1.


1. Найти численное и приближенное решение в виде степенного ряда до 6-ого порядка задачи Коши: $y'' - x \sin(y) = \sin 2x$, $y(0) = 0$, $y'(0) = 1$.

Сначала найдем численное решение задачи Коши и построим его график.

```
> restart; Ordev=6:
> eq:=diff(y(x), x$2) - x*sin(y(x)) = sin(2*x) :
> cond:=y(0)=0, D(y)(0)=1:
> de:=dsolve({eq, cond}, y(x), numeric) ;
de:=proc(rkf45_x)...end
```

Замечание: в строке вывода появляется сообщение о том, что при решении использован метод *rkf45*. Во избежание вывода строк, не несущих полезной информации, рекомендуется отделять промежуточные команды двоеточием. Если необходимо получить значение решения при каком-то фиксированном значении переменной x (заодно будет выведено значение производной решения в этой точке), например, при $x=0.5$, то следует набрать:

```
> de(0.5) ;
[ x = .5, y(x) = .5449261153862630, ∂/∂x y(x) = 1.272503082225380 ]
> with(plots) :
> odeplot(de, [x, y(x)], -10..10, thickness=2) ;
```




Теперь найдем приближенное решение задачи Коши в виде степенного ряда и построим графики численного решения и полученного степенного ряда в интервале их наилучшего совпадения.

```
> dsolve({eq, cond}, y(x), series)
```

$$y(x) = x + \frac{1}{3}x^3 + \frac{1}{12}x^4 - \frac{1}{15}x^5 + O(x^6)$$

```
> convert(%, polynom):p:=rhs(%):
> p1:=odeplot(de, [x, y(x)], -2..3, thickness=2,
  color=black):
> p2:=plot(p, x=-2..3, thickness=2, linestyle=3,
  color=blue):
> display(p1, p2);
```



Наилучшее приближение решения степенным рядом достигается примерно на интервале $-1 < x < 1$ (так же как и в примере 3 задания 1.5).


2. Построить графики решений задачи Коши системы дифференциальных уравнений:

$$x'(t) = 2y(t)\sin(t) - x(t) - t,$$

$$y'(t) = x(t),$$

$$x(0) = 1, y(0) = 2.$$

```
> restart; cond:=x(0)=1, y(0)=2:
> sys:=diff(x(t), t)=2*y(t)*sin(t)-x(t)-t,
  diff(y(t), t)=x(t):
> F:=dsolve({sys, cond}, [x(t), y(t)], numeric):
> with(plots):
> p1:=odeplot(F, [t, x(t)], -3..7, color=black,
  thickness=2, linestyle=3):
> p2:=odeplot(F, [t, y(t)], -3..7, color=green,
  thickness=2):
> p3:=textplot([3.5, 8, "x(t)", font=[TIMES,
  ITALIC, 12]]):
> p4:=textplot([5, 13, "y(t)", font=[TIMES,
  ITALIC, 12]]):
> display(p1, p2, p3, p4);
```



Пакет графического представления решений дифференциальных уравнений DEtools.

Для численного решения задачи Коши, построения графиков решения и фазовых портретов в *Maple* имеется специальный пакет **DEtools**.

Команда **DEplot** из пакета **DEtools** строит численными методами графики решения или фазовые портреты. Эта команда аналогична команде **odeplot**, но более функциональна. Она, в отличие от **odeplot**, сама производит численное решение дифференциального уравнения. Основные параметры **DEplot** похожи на параметры **odeplot**: **DEplot**(**de**, **vars**, **range**, **x=x1..x2**, **y=y1..y2**, **cond**, **ptions**), где **de** – дифференциальное уравнение или система дифференциальных уравнений; **vars** – список неизвестных функций; **range** – диапазон измерения независимой переменной; **cond** – начальные условия; **x=x1..x2** и **y=y1..y2** – диапазоны изменения функций; **options** – дополнительные параметры.

Наиболее часто используемые параметры: **linecolor**=цвет линии; **scene=[x,y]** – определяет, какие зависимости выводить на график; **iterations**=число итераций, необходимое для повышения точности вычислений (по умолчанию это число равно 1); **stepsize**=число, равное расстоянию между точками на графике, по умолчанию оно равно $(x_2-x_1)/20$, этот параметр необходим для вывода более гладкой кривой решения; **obsrange=true/false** – прерывать или нет вычисления, если график решения выходит за установленный для рисования интервал.


Для решения дифференциального уравнения n -ого порядка начальные условия можно задавать в более компактной форме: $[x_0, y_0, y'_0, y''_0, \dots]$, где x_0 – точка, в которой задаются начальные условия, y_0 – значение искомой функции в точке x_0 , y'_0, y''_0, \dots – значения производных первой, второй и т.д. до $(n-1)$ -ого порядка.

Задание 2.2.

Нарисовать график решения дифференциального уравнения:

$$y''' + x\sqrt{|y'|} + x^2 y = 0, \quad y(0) = 0, \quad y'(0) = 1, \quad y''(0) = 1 \quad \text{в интервале} \\ x \in [-4, 5].$$

```
> restart; with(DEtools):
> DEplot(diff(y(x), x$3) + x*sqrt(abs(diff(y(x), x)))
+ x^2*y(x)=0, {y(x)}, =-4..5, [[y(0)=0, D(y)(0)=1,
(D@@2)(y)(0)=1]], stepsize=.1, linecolor=black,
thickness=2);
```



Построение фазовых портретов систем дифференциальных уравнений.

Для дифференциального уравнения порядка выше первого команда **DEplot** рисует только кривые решений дифференциальных уравнений, а для систем дифференциальных уравнений первого порядка могут быть нарисованы и фазовые портреты.

С помощью команды **DEplot** можно построить фазовый портрет в плоскости (x, y) , для системы двух дифференциальных уравнений:

$$\frac{dx}{dt} = f(x, y, t), \quad \frac{dy}{dt} = g(x, y, t), \quad \text{если в параметрах данной команды}$$

указать **scene=[x, y]**.

Если система дифференциальных уравнений является автономной, то на фазовом портрете будет построено поле

направлений в виде стрелок. Размер стрелок регулируется параметром **arrows=SMALL, MEDIUM, LARGE, LINE** или **NONE**.

Для того, чтобы нарисовать весь фазовый портрет, необходимо для каждой фазовой траектории указывать начальные условия: например, для системы двух дифференциальных уравнений первого порядка несколько начальных условий в команде **DEplots** указываются после задания диапазона изменения независимой переменной **t**: **[[x(0)=x1, y(0)=y1], [x(0)=x2, y(0)=y2], ..., [x(0)=xn, y(0)=yn]]**.

Начальные условия можно задавать в более компактной форме: **[t0, x0, y0]**, где **t0** – точка, в которой задаются начальные условия, **x0** и **y0** – значения искомых функций в точке **t0**.

Фазовый портрет системы двух дифференциальных уравнений первого порядка можно также построить с помощью команды **phaseportrait(sys, [x,y], x1..x2, [[cond]])**, где **sys** – система двух дифференциальных уравнений первого порядка, **[x,y]** – имена искомых функций, **x1..x2** – интервал, на котором следует построить фазовый портрет, а в фигурных скобках указываются начальные условия. Эта команда находится в пакете **DEtools**, поэтому данный пакет должен быть предварительно загружен.


Задание 2.3.

1. Построить фазовый портрет системы дифференциальных

уравнений:
$$\begin{cases} x' = y \\ y' = x - x^3 \end{cases}$$

для нескольких наборов начальных условий: $x(0)=1, y(0)=0.2$; $x(0)=0, y(0)=1$; $x(0)=1, y(0)=0.4$; $x(0)=1, y(0)=0.75$; $x(0)=0, y(0)=1.5$; $x(0)=-0.1, y(0)=0.7$.


```
> restart; with(DEtools):
> DEplot({diff(x(t),t)=y, diff(y(t),t)=x-x^3},
  [x(t),y(t)], t=0..20, [[0,1,0.2], [0,0,1],
  [0,1,0.4], [0,1,0.75], [0,0,1.5], [0,-0.1,0.7]],
  stepsize=0.1, arrows=none, linecolor=black);
```

2. Построить фазовый портрет с полем направлений автономной системы $\begin{cases} x' = y \\ y' = \sin x \end{cases}$

для различных начальных условий $x(0)=1, y(0)=0$; $x(0)=-1, y(0)=0$; $x(0)=\pi, y(0)=1$; $x(0)=-\pi, y(0)=1$; $x(0)=3\pi, y(0)=0.2$; $x(0)=3\pi, y(0)=1$; $x(0)=3\pi, y(0)=1.8$; $x(0)=-2\pi, y(0)=1$.


```
> restart; with(DEtools):
> sys:=diff(x(t),t)=y, diff(y(t),t)=sin(x):
> DEplot({sys},[x(t),y(t)], t=0..4*Pi, [[0,1,0],
[0,-1,0], [0,Pi,1], [0,-Pi,1], [0,3*Pi,0.2],
[0,3*Pi,1], [0,3*Pi,1.8], [0,-2*Pi,1]],
stepsize=0.1, linecolor=black);
```



3. Построить фазовый портрет системы дифференциальных уравнений: $\begin{cases} x' = 3x + y \\ y' = y - x \end{cases}$

Начальные условия, диапазон изменения переменной и размеры координатных осей подбираются самостоятельно из соображений наглядности фазового портрета.

```
> restart; with(DEtools):
> sys:=diff(x(t),t)=3*x+y, diff(y(t),t)=-x+y:
> phaseportrait([sys],[x(t),y(t)],t=-10..10,
[[0,1,-2],[0,-3,-3],[0,-2,4],[0,5,5],[0,5,-3],
[0,-5,2],[0,5,2],[0,-1,2]],x=-30..30,y=-20..20,
stepsize=.1, colour=blue, linecolor=black);
```



Контрольные задания.

1. Найти общее решение дифференциального уравнения:

$$y'' - 2y' - 3y = xe^{4x} \sin x$$

2. Найти фундаментальную систему решений дифференциального уравнения:

$$y''' + y'' = 1 - 6x^2 e^{-x}$$

3. Найти решение задачи Коши: $y''' - y' = \operatorname{tg} x$, $y(0) = 3$, $y'(0) = -1$, $y''(0) = 1$

4. Найти решение системы дифференциальных уравнений:

$$\begin{cases} x'' + 5x' + 2y' + y = 0 \\ 3x'' + 5x + y' + 3y = 0 \end{cases}$$

при начальных условиях $x(0)=1$, $x'(0)=0$; $y(0)=1$.

5. Найти решение нелинейного уравнения $y'' + y = y^2$ при начальных условиях $y(0)=2a$, $y'(0)=a$ в виде разложения в степенной ряд до 6-го порядка.

6. Построить график численного решения задачи Коши $y' = \sin(xy)$, $y(0)=1$.

7. Решить численно задачу Коши: $y'' = xy' - y^2$, $y(0) = 1$, $y'(0) = 2$.
Найти приближенное решение этого уравнения в виде разложения в

степенной ряд. Построить на одном рисунке графики полученных решений.

8. Построить график численного решения задачи Коши $y'' - xy' + xy = 0$, $y(0)=1$, $y'(0)=-4$ на интервале $[-1.5; 3]$, используя команду **DEplot**.

9. Построить фазовый портрет системы дифференциальных уравнений

$$\begin{cases} x' = 3x - 4y \\ y' = x - 2y \end{cases}$$

при нескольких начальных условиях, которые следует подобрать самостоятельно для наилучшей наглядности рисунка.

Контрольные вопросы.

1. Какая команда позволяет решить дифференциальное уравнение? Опишите ее параметры.
2. С помощью каких операторов обозначается производная в дифференциальном уравнении и в начальных условиях?
3. Какой параметр команды **dsolve** следует установить, чтобы получить фундаментальную систему дифференциальных уравнений?
4. Какой параметр команды **dsolve** следует установить, чтобы получить приближенное решение дифференциального уравнения в виде разложения в степенной ряд? Как определяется порядок разложения?
5. Опишите, какие команды нужно ввести, прежде чем построить график приближенного решения, полученного в виде степенного ряда.
6. Какой параметр команды **dsolve** следует установить, чтобы решить дифференциальное уравнение численно?
7. Как найти значение решения дифференциального уравнения в какой-либо конкретной точке?
8. Какая команда позволяет построить график численно решенного дифференциального уравнения? В каком пакете находится эта команда?
9. Какой пакет предназначен для графического представления и численного решения дифференциального уравнения?
10. В чем отличие команд **odeplot** и **DEplot**?
11. Опишите способы построения фазового портрета системы дифференциальных уравнений.

VII. Математический анализ: функции многих переменных, векторный анализ, ряды, интегральные преобразования

1. Дифференциальное исчисление функций многих переменных.
2. Интегральное исчисление функций многих переменных.
3. Векторный анализ.
4. Ряды и произведения.
5. Интегральные преобразования.

§1. Дифференциальное исчисление функций многих переменных

Большинство задач дифференциального и интегрального исчисления функций многих переменных решается в *Maple* теми же командами, что и для функций одной переменной, только с указанием дополнительных параметров.

Частные производные.

Для вычисления частных производных функции $f(x_1, \dots, x_m)$ используется уже хорошо известная вам команда **diff**. В этом случае эта команда имеет такой формат: **diff (f, x1\$н1, x2\$н2, ..., xm\$nm)**, где **x1, ..., xm** – переменные, по которым производится дифференцирование, а после знака **\$** указаны соответствующие порядки дифференцирования. Например, частная производная $\frac{\partial^2 f}{\partial x \partial y}$ записывается в виде: **diff (f, x, y)**.

Задание 1.1.

1. Найти $\frac{\partial f}{\partial x}$ и $\frac{\partial f}{\partial y}$ функции $f = \arctg \frac{x}{y}$.

> **f:=arctan (x/y) :**

D> **iff (f, x)=simplify (diff (f, x)) ;**

$$\frac{\partial}{\partial x} \arctan \frac{x}{y} = \frac{y}{x^2 + y^2}$$

> **Diff (f, y)=simplify (diff (f, y)) ;**

$$\frac{\partial}{\partial y} \arctan \frac{x}{y} = -\frac{x}{x^2 + y^2}.$$

2. Найти все частные производные 2-го порядка функции

$$f(x, y) = \frac{x - y}{x + y}.$$

> **restart; f:=(x-y)/(x+y);**

> **Diff(f, x\$2)=simplify(diff(f, x\$2));**

$$\frac{\partial^2}{\partial x^2} \frac{x - y}{x + y} = -4 \frac{y}{(x + y)^3}$$

> **Diff(f, y\$2)=simplify(diff(f, y\$2));**

$$\frac{\partial^2}{\partial y^2} \frac{x - y}{x + y} = 4 \frac{x}{(x + y)^3}$$

> **Diff(f, x, y)=diff(f, x, y);**

$$\frac{\partial^2}{\partial x \partial y} \frac{x - y}{x + y} = 2 \frac{x - y}{(x + y)^3}.$$

Локальные и условные экстремумы функций многих переменных.

Для исследования функции на локальный и условный экстремум используется команда из стандартной библиотеки **extrema(f, {cond}, {x, y, ...}, 's')**, где **cond** – ограничения для поиска условного экстремума, которые записываются в виде равенств. После ограничений в фигурных скобках указываются все переменные, от которых зависит функция **f**, а затем в кавычках записывается **s** – имя переменной, которой будут присвоены координаты точек экстремума. Если ограничений не указывать, то будет производиться поиск локального экстремума.

К сожалению, команда **extrema** выдает все критические точки, то есть и те, в которых экстремума нет. Отсеять ненаходящие экстремума критические точки можно с помощью непосредственной подстановки этих точек в функцию, например, оператором **subs**.

Так же, как и для функции одной переменной, наибольшее и наименьшее значения функции многих переменных вычисляются командами **maximize(f, {x1, ..., xn}, range)**, и **minimize(f, {x1, ..., xn}, range)**, где следует указывать после функции в фигурных скобках список всех переменных, от которых она

зависит, а затем интервалы для каждой переменной, указывающие область поиска наибольшего и наименьшего значений.

Если требуется найти переменные, при которых линейная функция многих переменных имеет максимум (или минимум) при выполнении определенных ограничений, заданных в виде линейных равенств или неравенств, то следует использовать симплекс-метод. Для этого сначала необходимо загрузит пакет **simplex**, а затем воспользоваться командой **maximize** (или **minimize**), где теперь в качестве **range** можно указывать в фигурных скобках ограничительную систему неравенств. Пакет **simplex** предназначен для решения задач линейной оптимизации. После его загрузки команды **maximize** и **minimize** меняют свое действие. Теперь эти команды выдают координаты точек, при которых заданная линейная функция имеет максимум или минимум. При этом допускается дополнительная опция для поиска только неотрицательных решений **NONNEGATIVE**.

Задание 1.2.

1. Найти экстремумы функции $f(x, y) = 2x^4 + y^4 - x^2 - 2y^2$.

```
> restart: readlib(extrema):
> f:=2*x^4+y^4-x^2-2*y^2:
> extrema(f, {}, {x,y}, 's');s;
```

$$\left\{0, \frac{-9}{8}\right\}$$

$$\left\{\{x=0, y=0\}, \left\{x=\frac{1}{2}, y=0\right\}, \left\{x=-\frac{1}{2}, y=0\right\}, \{x=0, y=1\}, \{x=0, y=-1\},$$

$$\left\{x=\frac{1}{2}, y=1\right\}, \left\{x=\frac{1}{2}, y=-1\right\}, \left\{x=-\frac{1}{2}, y=1\right\}, \left\{x=-\frac{1}{2}, y=-1\right\}\right\}$$

Получилось всего два экстремума, поэтому очевидно, что $f_{\max}=0$ и $f_{\min}=-9/8$, причем максимум достигается в точке $(0,0)$. Остальные критические точки следует проверить. В силу четности функции по обоим переменным, можно ограничиться проверкой критических точек только с положительными координатами.

```
> subs ([x=1/2, y=1], f);
```

$$\frac{-9}{8}$$

```
> subs ([x=1/2, y=0], f);
```

$$\frac{-1}{8}$$

> subs ([x=0, y=1], f);

-1

Таким образом, функция имеет следующие локальные экстремумы: $f_{\max}=f(0,0)=0$ и $f_{\min}=f\left(\pm\frac{1}{2}, \pm 1\right)=f\left(\pm\frac{1}{2}, \mp 1\right)=-9/8$.

2. Найти наибольшее и наименьшее значения функции $f(x, y) = x^2 + 2xy - 4x + 8y$ в прямоугольнике $x=0, y=0, x=1, y=2$.

Замечание: заданную область удобнее записывать в виде неравенств: $0 < x < 1, 0 < y < 2$.

> restart: readlib(maximize): readlib(minimize):

> f:=x^2+2*x*y-4*x+8*y:

> maximize(f, {x, y}, {x=0..1, y=0..2});

17

> minimize(f, {x, y}, {x=0..1, y=0..2});

-4

Таким образом, функция имеет наибольшее значение $f_{\max}=17$ и наименьшее значение $f_{\min}=-4$.

3. Найти условные экстремумы функции $f(x, y, z) = xy + yz$ при $x^2 + y^2 = 2, y + z = 2, x > 0, y > 0, z > 0$.

> restart: readlib(extrema): f:=x*y+y*z:

> assume(x>0); assume(y>0); assume(z>0);

> simplify(extrema(f, {x^2+y^2=2, y+z=2}, {x, y, z}, 's'));

$$\left\{ \min\left(\frac{3}{2} \text{RootOf}(-Z^2+4Z+1) + \frac{1}{2}, 0\right), \max\left(\frac{3}{2} \text{RootOf}(-Z^2+4Z+1) + \frac{1}{2}, 2\right) \right\}$$

Несмотря на предварительное использование команды упрощения выражения **simplify**, полученный результат имеет не аналитический вид, однако это можно исправить, если воспользоваться командой **convert**.

> convert(%, radical);

$$\left\{ \min\left(-\frac{5}{2} + \frac{3}{2}\sqrt{3}, 0\right), \max\left(-\frac{5}{2} + \frac{3}{2}\sqrt{3}, 2\right) \right\}$$

> convert(s, radical);

$$\left\{ \{x=1, z=1, y=1\}, \{x=-1, z=1, y=1\}, \left\{x=-\frac{1}{2} - \frac{1}{2}\sqrt{3}, y=-\frac{1}{2}(-2 + \sqrt{3})(1 + \sqrt{3}), z=-\frac{5}{2} - \frac{1}{2}\sqrt{3}\right\} \right\}$$

В этом случае команда **extrema** сама определила характер экстремумов, однако, в каких точках функция имеет экстремумы, можно определить подстановкой.

```
> subs (s [1] , f) ;
2
> subs (s [2] , f) ;
0
> subs (s [3] , f) : convert (% , radical) : simplify (%) ;
- 5/2 + 3/2 sqrt(3)
```

Таким образом, функция имеет следующие условные экстремумы: $f_{\max}=f(1,1,1)=2$ и $f_{\min}=f(-1,1,1)=0$; третья критическая точка является седловой.

4. При каких значениях переменных функция $f(x,y,z)=-x+2y+3z$ имеет максимум, если требуется выполнение условий $x+2y-3z \leq 4$, $5x-6y+7z \leq 8$, $9x+10z \leq 11$, а все переменные неотрицательные?

```
> restart: with(simplex):
Warning, new definition for maximize
Warning, new definition for minimize
> f:=-x+2*y+3*z:
> cond:={x+2*y-3*z<=4, 5*x-6*y+7*z<=8,
9*x+10*z<=11}:
> maximize (f , cond, NONNEGATIVE ) ;
```

$$\left\{ x=0, y=\frac{73}{20}, z=\frac{11}{10} \right\}$$

§2 Интегральное исчисление функций многих переменных

В *Maple* имеются две специальные команды для вычисления двойных и тройных интегралов, содержащиеся в библиотеке **student**.

Для вычисления двойных интегралов $\iint_D f(x,y) dx dy$ используется

команда **Doubleint (f(x, y), D)**, где **D** – область интегрирования, записываемая в одном из следующих форматов:

- **x=x1..x2, y=y1..y2**, где числа **x1, x2, y1, y2** задают прямоугольную область интегрирования;

- $\mathbf{x=f1(y) .. f2(y)}$, $\mathbf{y=y1 .. y2}$, где $\mathbf{f1(y)}$, $\mathbf{f2(y)}$ – линии, ограничивающие область интегрирования слева и справа на интервале от $\mathbf{y1}$ до $\mathbf{y2}$;
- $\mathbf{x=x1 .. x2}$, $\mathbf{y=g1(x) .. g2(x)}$, где $\mathbf{g1(y)}$, $\mathbf{g2(y)}$ – линии, ограничивающие область интегрирования снизу и сверху на интервале от $\mathbf{x1}$ до $\mathbf{x2}$.

Для вычисления тройных интегралов
$$\iiint_V f(x,y,z) dx dy dz$$

используется команда **Tripleint(f(x, y, z), x, y, z, V)** , где **V** – область интегрирования.

Обе эти команды являются командами отложенного действия. Чтобы получить значение интеграла, следует использовать команду **value(%)**.

Повторные интегралы можно вычислять с помощью повторения

команды **int**, например, повторный интеграл
$$\int_0^2 dy \int_0^1 x^2 y^3 dx$$

вычисляется командой

```
> int(int(x^2*y^3, x=0..1), y=0..2);
```

$$\frac{4}{3}$$

Задание 2.

1. Вычислить повторный интеграл
$$\int_2^4 dy \int_0^y \frac{y^3}{x^2 + y^2} dx$$

```
> Int(Int(y^3/(x^2+y^2), x=0..y), y=2..4) =
int(int(y^3/(x^2+y^2), x=0..y), y=2..4);
```

$$\int_2^4 dy \int_0^y \frac{y^3}{x^2 + y^2} dx = \frac{14}{3} \pi$$

2. Вычислить двойной интеграл
$$\iint_D \sin(x+2y) dx dy$$
 по области,

ограниченной линиями $y = 0$, $y = x$, $x + y = \frac{\pi}{2}$.

Замечание: сначала следует описать область интегрирования D в виде неравенств: $D = \{(x, y) : y \leq x \leq \frac{\pi}{2} - y, 0 \leq y \leq \frac{\pi}{2}\}$

> **restart: with(student):**

> **J:=Doubleint(sin(x+2*y), x=y..Pi/2-y, y=0..Pi/2);**

$$J := \int_0^{\frac{1}{2}\pi} \int_y^{\frac{1}{2}\pi-y} \sin(x+2y) dx dy$$

> **J:=value(%);**

$$J := \frac{2}{3}$$

3. Вычислить тройной интеграл $\int_{-1}^1 dx \int_{x^2}^1 dy \int_0^2 (4+z) dz$.

Замечание: следует помнить, что порядок интегрирования определяется последовательностью пределов, поэтому сначала внутренние указываются пределы, содержащие функции.

> **J:=Tripleint(4+z, y=x^2..1, x=-1..1, z=0..2);**

$$J := \int_0^2 \int_{-1}^1 \int_{x^2}^1 4 + z dy dx dz$$

> **J:=value(%);**

$$J := \frac{40}{3}$$

§3. Векторный анализ

Приведем определения основных дифференциальных операций векторного анализа и команды *Maple* для их вычисления, которые содержатся в библиотеке **linalg**.

Градиент скалярной функции $f(x, y, z)$ – это вектор, координатами которого являются частные производные по соответствующим переменным: $\text{grad } f(x, y, z) = \left(\frac{\partial f}{\partial x}, \frac{\partial f}{\partial y}, \frac{\partial f}{\partial z} \right)$. В *Maple* **grad** вычисляется одноименной командой **grad(f, [x, y, z], c)**, где здесь и в

дальнейшем \mathbf{f} – функция, $[\mathbf{x}, \mathbf{y}, \mathbf{z}]$ – набор переменных, от которых она зависит.

Параметр \mathbf{c} позволяет вычислять данную дифференциальную операцию в различных криволинейных координатах (по умолчанию используется прямоугольная декартова система координат). Этот параметр может указываться во всех имеющихся в *Maple* дифференциальных операциях. Для вычисления дифференциальной операции в цилиндрических координатах следует записать **coords=cylindrical**, в сферических координатах – **coords=spherical**.

Лапласиан скалярной функции $f(x, y, z)$ – это оператор, действующий на функцию $f(x, y, z)$ по правилу:

$$\Delta f = \frac{\partial^2 f}{\partial x^2} + \frac{\partial^2 f}{\partial y^2} + \frac{\partial^2 f}{\partial z^2}. \quad \text{Он вычисляется командой}$$

laplacian(f, [x, y, z], c).

Дивергенцией вектор-функции $\mathbf{F}(x, y, z)$ называется функция (скалярная), вычисляемая по правилу: $\operatorname{div} \mathbf{F}(x, y, z) = \frac{\partial F_x}{\partial x} + \frac{\partial F_y}{\partial y} + \frac{\partial F_z}{\partial z}$.

Дивергенция в *Maple* вычисляется командой **diverge(F, [x, y, z], c)**, где здесь и в дальнейшем \mathbf{F} – вектор-функция, $[\mathbf{x}, \mathbf{y}, \mathbf{z}]$ – набор переменных, от которых она зависит.

Ротором вектор-функции $\mathbf{F}(x, y, z)$ называется вектор с координатами:

$$\operatorname{rot} \mathbf{F} = \left[\left(\frac{\partial F_z}{\partial y} - \frac{\partial F_y}{\partial z} \right), \left(\frac{\partial F_x}{\partial z} - \frac{\partial F_z}{\partial x} \right), \left(\frac{\partial F_y}{\partial x} - \frac{\partial F_x}{\partial y} \right) \right].$$

Ротор вычисляется командой **curl(F, [x, y, z], c)**.

Для вектор-функции $\mathbf{F}(x, y, z)$ можно вычислить матрицу Якоби

$$J = \begin{bmatrix} \frac{\partial F_x}{\partial x} & \frac{\partial F_y}{\partial x} & \frac{\partial F_z}{\partial x} \\ \frac{\partial F_x}{\partial y} & \frac{\partial F_y}{\partial y} & \frac{\partial F_z}{\partial y} \\ \frac{\partial F_x}{\partial z} & \frac{\partial F_y}{\partial z} & \frac{\partial F_z}{\partial z} \end{bmatrix}$$

с помощью команды **jacobian(F, [x, y, z])**.

Задание 3.

1. Дана функция $u(x, y) = \operatorname{arctg} \frac{y}{x}$. Найти $\operatorname{grad} u(x, y)$. Какие углы составляет $\operatorname{grad} u$ с осями координат? Найти производную функции $u(x, y)$ по направлению вектора $q = [1, 1]$.

```
> restart: with(linalg):
Warning, new definition for norm
Warning, new definition for trace
> u:=arctan(y/x): g:=simplify(grad(u, [x, y]));
```

$$g := \left[-\frac{y}{x^2 + y^2}, \frac{x}{x^2 + y^2} \right]$$

```
> alpha:=simplify(angle(g, [1, 0]));
```

$$\alpha := \pi - \arccos \left(\frac{y}{(x^2 + y^2) \sqrt{\frac{1}{x^2 + y^2}}} \right)$$

```
> beta:=simplify(angle(g, [0, 1]));
```

$$\beta := \arccos \left(\frac{x}{(x^2 + y^2) \sqrt{\frac{1}{x^2 + y^2}}} \right)$$

Косинусы этих углов являются направляющими косинусами $\operatorname{grad} u(x, y)$. Убедимся, что сумма их квадратов равна единице.

```
> simplify(cos(alpha)^2+cos(beta)^2);
1
```

Производная функции u по направлению q равна скалярному произведению градиента этой функции на нормированный вектор q :

$\frac{\partial u}{\partial q} = (\operatorname{grad} u, e)$, где $e = \frac{q}{\|q\|}$ – нормированный вектор q .

```
> q:=vector([1, 1]); e:=normalize(q);
```

$$q := [1, 1]$$

$$e := \left[\frac{1}{2}\sqrt{2}, \frac{1}{2}\sqrt{2} \right]$$

```
> udq:=simplify(dotprod(g, e));
```

$$udq := \frac{1}{2} \frac{\sqrt{2}(-y+x)}{x^2+y^2}$$

2. Дана вектор-функция $F(x, y, z) = [x^2yz, xy^2z, xyz^2]$. Найти $\operatorname{div}F$ и $\operatorname{rot}F$.

```
> F:=vector([x^2*y*z, x*y^2*z, x*y*z^2]);
```

```
> divF:=diverge(F, [x, y, z]);
```

$$\operatorname{div}F := 6xyz$$

```
> rotF:=curl(F, [x, y, z]);
```

$$\operatorname{rot}F := [xz^2 - xy^2, x^2z - xz^2, y^2z - x^2z]$$

3. При каком значении параметра a функция $u = x^3 + axy^2$ удовлетворяет уравнению Лапласа $\Delta u = 0$?

```
> u:=x^3+a*x*y^2;
```

```
> Delta(u):=laplacian(u, [x, y]);
```

$$\Delta(x^3 + axy^2) := 6x + 2ax$$

```
> a=solve(%=0, a);
```

$$a = -3$$

4. Доказать, что функция $u = \frac{e^{-kr} + e^{kr}}{r}$, где $r = \sqrt{x^2 + y^2 + z^2}$

удовлетворяет дифференциальному уравнению $\Delta u - k^2u = 0$, k – постоянная.

```
> u:=(exp(-k*r)+exp(k*r))/r;
```

```
> Delta(u):=simplify(laplacian(u, [r, theta, phi],
coords=spherical));
```

$$\Delta \left(\frac{e^{(-kr)} + e^{(kr)}}{r} \right) := \frac{k^2(e^{(-2kr)} + 1)e^{(kr)}}{r}$$

```
> simplify(%-k^2*u);
```

$$0$$

5. Найти матрицу Якоби и ее определитель вектор-функция $v = [x, y/x]$.

```
> v:=vector([x, y/x]): jacobian(v, [x, y]);
```

$$\begin{bmatrix} 1 & 0 \\ -\frac{y}{x^2} & \frac{1}{x} \end{bmatrix}$$

```
> det(%);
```

$$\frac{1}{x}$$

§4. Ряды и произведения

Вычисление суммы ряда и произведений.

Конечные и бесконечные суммы $\sum_{n=a}^b S(n)$ вычисляются командой прямого исполнения **sum** и отложенного исполнения **Sum**. Аргументы этих команд одинаковые: **sum(expr, n=a..b)**, где **expr** – выражение, зависящее от индекса суммирования, **a..b** – пределы индекса суммирования, указывающие, что суммировать следует от **n=a** до **n=b**.

Если требуется вычислить сумму бесконечного ряда, то в качестве верхнего предела вводится **infinity**.

Аналогичным образом вычисляются произведения $\prod_{n=a}^b P(n)$ командами прямого **product(P(n), n=a..b)** и отложенного действий **Product P(n), n=a..b**.

Задание 4.1.

1. Найти полную и *N*-частичную суммы ряда, общий член которого

равен: $a_n = \frac{1}{(3n-2)(3n+1)}$.

> **restart: a[n] := 1 / ((3*n-2) * (3*n+1));**

$$a_n := \frac{1}{(3n-2)(3n+1)}$$

> **S[N] := Sum(a[n], n=1..N) = sum(a[n], n=1..N);**

$$S_N := \sum_{n=1}^N \frac{1}{(3n-2)(3n+1)} = -\frac{1}{3} \frac{1}{3N+1} + \frac{1}{3}$$

> **S := limit(rhs(S[N]), N=+infinity);**

$$S := \frac{1}{3}$$

2. К какой функции сходится степенной ряд: $\sum_{n=1}^{\infty} (-1)^{n+1} n^2 x^n$?

> **Sum((-1)^(n+1)*n^2*x^n, n=1..infinity) =
sum((-1)^(n+1)*n^2*x^n, n=1..infinity);**

$$\sum_{n=1}^{\infty} (-1)^{n+1} n^2 x^n = \frac{x(-x+1)}{(x+1)^3}.$$

3. Найти сумму степенного ряда $\sum_{n=0}^{\infty} \frac{(1+x)^n}{(n+1)n!}$.

> **Sum((1+x)^n/((n+1)*n!), n=0..infinity) =
sum((1+x)^n/((n+1)*n!), n=0..infinity);**

$$\sum_{n=0}^{\infty} \frac{(1+x)^n}{(n+1)n!} = \frac{e^{(x+1)}(1-e^{-(x-1)})}{x+1}$$

4. Найти сумму биномиального ряда $\sum_{n=1}^{\infty} C_n^4 (1-x)^n$.

> **Sum(binomial(n,4)*(1-x)^n, n=1..infinity) =
sum(binomial(n,4)*(1-x)^n, n=1..infinity);**

$$\sum_{n=1}^{\infty} \text{binomial}(n,4)(1-x)^n = \frac{(1-x)^4}{x^5}$$

5. Вычислить бесконечное произведение: $\prod_{n=2}^{\infty} \frac{n^3-1}{n^3+1}$

> **Product((n^3-1)/(n^3+1), n=2..infinity) =
product((n^3-1)/(n^3+1), n=2..infinity);**

$$\prod_{n=2}^{\infty} \frac{n^3-1}{n^3+1} = \frac{2}{3}.$$

Разложение функции в степенной ряд и ряд Тейлора.

Разложение функции $f(x)$ в степенной ряд в окрестности точки a

$$f(x) = C_0 + C_1(x-a) + C_2(x-a)^2 + \dots + O(x^n)$$

осуществляется командой **series(f(x), x=a, n)**, где a – точка, в окрестности которой производится разложение, n – число членов ряда.

Аналогичного действия команда **taylor(f(x), x=a, n)** раскладывает функции $f(x)$ в окрестности точки $x=a$ до порядка $n-1$ по формуле Тейлора.

Команды **series** и **taylor** выдают результат, имеющий тип series. Для того, чтобы иметь возможность дальнейшей работы с

полученным разложением, его следует преобразовать в полином с помощью команды **convert(% , polynom)** .

Функцию многих переменных $f(x_1, \dots, x_n)$ можно разложить в ряд Тейлора по набору переменных (x_1, \dots, x_n) в окрестности точки (a_1, \dots, a_n) до порядка n с помощью команды **mtaylor(f(x), [x1, ..., xn], n)** . Эта команда находится в стандартной библиотеке, поэтому перед использованием должна быть вызвана **readlib(mtaylor)** .

Задание 4.2.

1. Разложить в степенной ряд $f(x) = e^{-x} \sqrt{x+1}$ в окрестности $x_0=0$, удерживая 5 первых членов.

> **f(x)=series(exp(-x)*sqrt(x+1), x=0, 5);**

$$f(x) = 1 - \frac{1}{2}x - \frac{1}{8}x^2 + \frac{13}{48}x^3 - \frac{79}{384}x^4 + O(x^5)$$


2. Построить на одном рисунке графики интеграла ошибок

$\text{erf}(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-t^2} dt$ и его разложения в ряд Тейлора в окрестности нуля.

> **taylor(erf(x), x, 8) : p:=convert(% , polynom);**

$$p := 2 \frac{x}{\sqrt{\pi}} - \frac{2}{3} \frac{x^3}{\sqrt{\pi}} + \frac{1}{5} \frac{x^5}{\sqrt{\pi}} - \frac{1}{21} \frac{x^7}{\sqrt{\pi}}$$

> **plot({erf(x), p}, x=-2..2, thickness=[2,2], linestyle=[1,3], color=[red,green]);**



Пунктирной линией изображен график ряда Тейлора, а сплошной – самой функции.

3. Разложить $f(x, y) = \sin(x^2 + y^2)$ в ряд Тейлора в окрестности точки $(0, 0)$ до 6-ого порядка.

> **readlib(mtaylor) :**


```
> f:=mtaylor(sin(x^2+y^2), [x=0,y=0], 7);
```

$$f = x^2 + y^2 - \frac{1}{6}x^6 - \frac{1}{2}y^2x^4 - \frac{1}{2}y^4x^2 - \frac{1}{6}y^6$$

Создание собственных процедур. Разложение функции в ряд Фурье.

В *Maple* имеется возможность создавать собственные процедуры. Процедура начинается с заголовка. Заголовок состоит из имени процедуры (его пользователь определяет сам), далее следует обязательный оператор присваивания := и служебное слово **proc**, после которого в круглых скобках через запятую указываются формальные параметры процедуры.

Во избежании неполадок работы процедуры, рекомендуется в строке заголовка процедуры описывать переменные, которые будут использоваться только внутри тела процедуры (они называются локальными переменными). Для этого используется служебное слово **local**, после которого через запятую перечисляются локальные переменные.

После заголовка следует основное тело процедуры, состоящее из составленных пользователем команд, причем последняя команда будет выводить окончательный результат выполнения процедуры. Процедура должна обязательно оканчиваться служебным словом **end**.

Общий вид процедуры (стандартный синтаксис):

```
> name:=proc(var1, var2, ...) local vloc1, vloc2, ...;
> expr1;
> expr2;
.....
> exprn;
> end;
```

В *Maple* нет команды, позволяющей производить разложение функции в тригонометрический ряд Фурье. Однако можно создать собственную процедуру разложения ряд Фурье. Пусть требуется разложить на интервале $[x_1, x_2]$ $2l$ -периодическую функцию $f(x)$. Тогда ряд Фурье имеет вид:

$$f(x) = \frac{a_0}{2} + \sum_{k=1}^{\infty} a_k \cos \frac{k\pi x}{l} + \sum_{k=1}^{\infty} b_k \sin \frac{k\pi x}{l},$$

где $l=(x_2-x_1)/2$;

$$a_0 = \frac{1}{l} \int_{x_1}^{x_2} f(x) dx; \quad a_k = \frac{1}{l} \int_{x_1}^{x_2} f(x) \cos \frac{k\pi x}{l} dx; \quad b_k = \frac{1}{l} \int_{x_1}^{x_2} f(x) \sin \frac{k\pi x}{l} dx.$$

Получить первые n членов ряда Фурье можно с помощью следующей процедуры:

```
> fourierseries:=proc(f,x,x1,x2,n) local k, l,
  a, b, s;
> l:=(x2-x1)/2;
> a[0]:=int(f,x=x1..x2)/l;
> a[k]:=int(f*cos(k*Pi*x/l),x=x1..x2)/l;
> b[k]:=int(f*sin(k*Pi*x/l),x=x1..x2)/l;
> s:=a[0]/2+sum(a[k]*cos(k*Pi*x/l)+
  b[k]*sin(k*Pi*x/l), k=1..n);
> end;
```

Порядок обращения к этой процедуре такой: **fourierseries(f,x,x1,x2,n)**, где **f** – имя функции, разложение которой требуется найти, где **x** – имя независимой переменной, где **x1, x2** – интервал разложения, где **n** – число членов ряда.

Задание 4.3.


1. Разложить в ряд Фурье функцию $f(x)=x/2$ с периодом 2π на интервале $[0; 2\pi]$, удерживая 6 членов ряда. Построить на одном рисунке графики функции и ее n -частичной суммы ряда Фурье.

Сначала полностью наберите процедуру **fourierseries**, предложенную выше в теоретической части.

```
> f:=x/2:x1:=0:x2:=2*Pi:
> fr:=fourierseries(f,x,x1,x2,6);
```

$$fr := \frac{1}{2}\pi - \sin(x) - \frac{1}{2}\sin(2x) - \frac{1}{3}\sin(3x) - \frac{1}{4}\sin(4x) - \frac{1}{5}\sin(5x) - \frac{1}{6}\sin(6x)$$

```
> plot({fr,f}, x=x1..x2, color=[blue,black],
  thickness=2, linestyle=[3,1]);
```




Пунктирной линией изображен график n -частичной суммы ряда Фурье, а сплошной – самой функции. По виду n -частичной суммы ряда Фурье в данном примере легко установить общий вид этого ряда:

$$f(x) = \frac{\pi}{2} - \sum_{k=1}^{\infty} \frac{\sin kx}{k}.$$

2. Разложить несколько раз в ряд Фурье функцию $f(x) = e^{-x}$ с периодом 2π на интервале $[\pi; -\pi]$, удерживая 2, 4 и 8 членов ряда. Построить на одном рисунке графики функции и ее n -частичных сумм ряда Фурье.

```
> f:=exp(-x); x1:=-Pi; x2:=Pi;
> fr1:=fourierseries(f, x, x1, x2, 2);
> fr2:=fourierseries(f, x, x1, x2, 4);
> fr3:=fourierseries(f, x, x1, x2, 8);
> plot({f, fr1, fr2, fr3}, x=x1..x2, color=[black, blue,
green, red], thickness=2, linestyle=[1,3,2,2]);
```



Сплошной линией изображен график функции, пунктирными – графики n -частичных сумм ряда Фурье. Видно, что чем больше слагаемых ряда удерживать, тем ближе расположен график суммы ряда к графику самой функции.

§5. Интегральные преобразования

В *Maple* имеется пакет **inttrans**, в котором содержатся команды различных интегральных преобразований.

Преобразование Фурье.

Прямое преобразование Фурье функции $f(x)$ вычисляется по формуле

$$F(k) = \int_{-\infty}^{\infty} f(x)e^{-ikx} dx .$$

В *Maple* оно может быть найдено командой **fourier (f(x), x, k)**, где **x** – переменная, по которой производится преобразование, **k** – имя переменной, которое следует присвоить параметру преобразования.

Обратное преобразование Фурье задается формулой

$$f(x) = \frac{1}{2\pi} \int_{-\infty}^{\infty} F(k)e^{ikx} dk$$

и вычисляется командой **invfourier (F(k), k, x)**.

Описанное выше прямое и обратное преобразования Фурье называются комплексными и применяются в тех случаях, когда функция $f(x)$ задана на всей числовой оси. Если функция $f(x)$ задана только при $x>0$, то рекомендуется использовать синус- и косинус-преобразования Фурье.

Прямое и обратное синус-преобразования Фурье функции $f(x)$ определяются формулами

$$F(k) = \sqrt{\frac{2}{\pi}} \int_0^{\infty} f(x) \sin kx dx \text{ и } f(x) = \sqrt{\frac{2}{\pi}} \int_{-\infty}^{\infty} F(k) \sin kx dk .$$

Поскольку формулы синус-преобразования Фурье симметричны относительно замены x на k , то в *Maple* эти преобразования вычисляются одной командой, но с различным порядком указания параметров: **fouriersin (f(x), x, k)** – вычисляет прямое синус-преобразование Фурье; **fouriersin (F(k), k, x)** – вычисляет обратное синус-преобразование Фурье.

Аналогично, прямое и обратное косинус-преобразования Фурье функции $f(x)$ определяются формулами

$$F(k) = \sqrt{\frac{2}{\pi}} \int_0^{\infty} f(x) \cos kx dx \quad \text{и} \quad f(x) = \sqrt{\frac{2}{\pi}} \int_{-\infty}^{\infty} F(k) \cos kx dk .$$

В *Maple* эти преобразования вычисляются одной командой, но с различным порядком указания параметров: **fouriercos (f (x) , x , k)** – вычисляет прямое косинус-преобразование Фурье; **fouriercos (F (k) , k , x)** – вычисляет обратное косинус-преобразование Фурье.

Задание 5.1.

1. Для функции $f(x) = e^{-a|x|}$, $a > 0$ найти преобразование Фурье.

```
> restart; with(inttrans): assume(a>0):
> fourier(exp(-a*abs(x)), x, k);
```

$$2 \frac{a}{k^2 + a^2}$$

2. Для функции $F(k) = \frac{1}{k^2 - a^2}$, $a > 0$ найти обратное преобразование Фурье.

```
> invfourier(1/(k^2-a^2), k, x);
```

$$-\frac{1}{4} \frac{I(\text{Heaviside}(x) - \text{Heaviside}(-x))(-e^{(Ia \sim x)} + e^{(-Ia \sim x)})}{a \sim}$$

После обратного преобразования Фурье результат часто содержит специальные функции (см. Лабораторную работу №1). В данном примере в строке вывода появилась функция Хевисайда:

$$\text{Heaviside}(x) = \begin{cases} 1, & x > 0 \\ 0, & x < 0 \end{cases}$$

Результат выполнения обратного преобразования Фурье может иметь более компактный вид после применения команды **convert(%, trig)**.

```
> convert(%, trig);
```

$$-\frac{1}{2} \frac{(\text{Heaviside}(x) - \text{Heaviside}(-x)) \sin(a \sim x)}{a \sim}$$

3. Для функции $f(x) = e^{-ax} \sin bx$, $a > 0$ найти синус- и косинус-преобразования Фурье.

```
> f:=exp(-a*x)*sin(b*x):
> fouriercos(f, x, k);
```

$$\frac{\sqrt{2} \left(\frac{1}{2} \frac{k+b}{a^2 + (k+b)^2} + \frac{1}{2} \frac{b-k}{a^2 + (b-k)^2} \right)}{\sqrt{\pi}}$$

> **fouriercos (f, x, k) ;**

$$\frac{1}{2} \frac{\sqrt{2} a \sim \left(\frac{1}{a^2 + (b-k)^2} - \frac{1}{a^2 + (k+b)^2} \right)}{\sqrt{\pi}}$$

Преобразование Лапласа.

Преобразование Лапласа функции $f(x)$ (если оно существует) вычисляется по формуле:

$$F(p) = \int_0^{\infty} f(x) e^{-px} dx .$$

Получаемая функция $F(p)$ называется изображением.

В *Maple* это преобразование вычисляется командой **laplace (f (x) , x, p)** , где **x** – переменная, по которой производится преобразование, **p** – имя переменной, которое следует присвоить параметру преобразования.

Обратное преобразование Лапласа (называется оригиналом) вычисляется по формуле:

$$f(x) = \frac{1}{2\pi i} \int_{a-i\infty}^{a+i\infty} F(p) e^{px} dp .$$

Оригинал $f(x)$ (если он существует) может быть найден по изображению $F(p)$ командой **invlaplace (F (p) , p, x)** .

Задание 5.2.

1. Найти изображение функции $f(x) = \cos ax \operatorname{sh} bx$.

> **restart:with(inttrans) :**

> **F(p)=laplace(cos(a*x)*sinh(b*x), x, p) ;**

$$F(p) = \frac{1}{2} \frac{p-b}{(p-b)^2 + a^2} - \frac{1}{2} \frac{p+b}{(p+b)^2 + a^2} .$$

2. Найти оригинал Лапласа функции $F(p) = \frac{1}{p^2 + 2ap}$, $a > 0$.

> **assume(a>0) : invlaplace(1/(p^2+2*a*p), p, x) :**

> **combine(% , trig) ;**

$$\frac{1}{2} \frac{1 - e^{(-2a \sim x)}}{a \sim}$$

Контрольные задания.

1. Найти все частные производные 2 – ого порядка функции

$$f(x, y) = \operatorname{arctg} \frac{x + y}{1 - xy}.$$

2. Найти условный экстремум функции

$$f(x, y, z) = y^2 + 4z^2 - 4yz - 2xz - 2xy, \text{ при } 2x^2 + 3y^2 + 6z^2 = 1$$

3. Найти значения переменных, при которых линейная функция $f(x, y, z) = x + y + z$ имеет максимум, если требуется выполнение условий $x+y \leq 2, z \leq 1$.

4. Вычислить тройной интеграл:

$$\int_0^{e-1} dx \int_0^{e-x-1} dy \int_e^{x+y+e} \frac{\ln(z-x-y) dz}{(x-e)(x+y-e)}.$$

5. Дана функция $f = xy - z^2$. Найти $\operatorname{grad} f$ и производную от f в направлении биссектрисы 1-ого координатного угла.

6. Жидкость заполняет пространство, вращаясь вокруг оси Oz против часовой стрелки с постоянной угловой скоростью $\Omega = [0, 0, \omega]$. Найти $\operatorname{div} V, \operatorname{rot} V, \operatorname{div} w$, где V – скорость, равная $V = [\Omega, r]$; w – угловое ускорение, равное $w = [\Omega, V]$; $r = [x, y, z]$ – радиус-вектор.

7. При какой зависимости частоты ω от параметров a, b и c функция $u = \cos(ax + by - \omega t)$ удовлетворяет волновому уравнению

$$\Delta u = \frac{1}{c^2} \frac{\partial^2 u}{\partial t^2} ?$$

8. Показать, что функция $\frac{1}{r}$ удовлетворяет уравнению Лапласа в

сферических координатах, а функция $\ln \frac{1}{\rho}$ – в цилиндрических.

9. Найти матрицу Якоби и ее определитель для вектор-функции $F = (r \cos \varphi \sin \theta, r \sin \varphi \sin \theta, r \cos \theta)$.

10. Найти сумму ряда $\sum_{n=1}^{\infty} \frac{1}{n(n+1)(n+2)}$ и сумму первых N членов.
11. Найти функцию, к которой сходится степенной ряд $\sum_{n=1}^{\infty} n(n+1)x^n$.
12. Разложить в степенной ряд $f(x)=\arcsin x$ в окрестности $x=0$ до 9-ого порядка.
13. Разложить в ряд Тейлора функцию $f(x, y) = \operatorname{arctg} \frac{x-y}{1+xy}$ до 6-ого порядка в окрестности точки $(0, 0)$.
14. Разложить в ряд Фурье функцию $f(x) = \begin{cases} 6, & 0 < x < 2 \\ 3x, & 2 \leq x < 4 \end{cases}$ с периодом 4 на интервале $[0;4]$, удерживая 6 членов ряда. Построить на одном рисунке графики функции и ее n -частичной суммы ряда Фурье.
15. Найти преобразование Фурье функции $f(x)=e^{-ax^2}$, $a>0$.
16. Найти изображения Лапласа и построить их графики для следующих функций:
а) $\frac{\sin t}{t}$; б) $\frac{1-\cos 2t}{t} e^{-3t}$.
17. Найти оригинал Лапласа функции $F(p) = \frac{1}{(p-1)^2(p^2+1)}$ и построить его график.
18. Дана функция $f(x) = \int_0^{\infty} \frac{1-\cos xt}{x^2} dx$, найти ее изображение Лапласа.

Контрольные вопросы.

1. Опишите, как в *Maple* вычисляются частные производные.
2. Какие команды используются для вычисления двойных и тройных интегралов? Опишите их параметры.
3. Для чего предназначен пакет **simplex**? В чем отличие команд **maximize** и **minimize** этого пакета от обычных **maximize** и **minimize**?

4. Что называется градиентом функции $f(x)$? Как он вычисляется в *Maple*?
5. Какие команды вычисляют дивергенцию и ротор вектор-функции?
6. Как вычислить сумму или произведение в *Maple*?
7. Какие команды осуществляют разложение функции в степенные ряды?
8. Каким образом в *Maple* создаются собственные процедуры? Опишите ее синтаксис.
9. Какие интегральные преобразования можно вычислить в *Maple*? Опишите команды прямых и обратных преобразований.

Рекомендуемая литература

1. Дьяконов В.П. Maple 6: учебный курс. СПб.: Питер, 2001.
2. Дьяконов В.П. Математическая система Maple V R3/R4/R5. М.: Солон, 1998.
3. Манзон Б.М. Maple V Power Edition. М.: Филинь, 1998.
4. Говорухин В.Н., Цибулин В.Г. Введение в Maple V. Математический пакет для всех. М.: Мир, 1997.
5. Прохоров Г.В., Леденев М.А., Колбеев В.В. Пакет символьных вычислений Maple V. М.: Петит, 1997.
6. Бугров Я.С., Никольский С.М. Элементы линейной алгебры и аналитической геометрии. М.: Наука. 1989.
7. Бугров Я.С., Никольский С.М. Дифференциальное и интегральное исчисление. М.: Наука. 1989.
8. Бугров Я.С., Никольский С.М. Дифференциальные уравнения. Кратные интегралы. Ряды. Функции комплексного переменного. М.: Наука, 1989.
9. Ильин В.А., Позняк Э.Г. Аналитическая геометрия. М.: Наука. 1970.
10. Ильин В.А., Позняк Э.Г. Линейная алгебра. М.: Наука. 1970.
11. Никольский С.М. Курс математического анализа (2 т.). М.: Наука. 1991.
12. Эльсгольц Л.Э. Дифференциальные уравнения и вариационное исчисление. М.: Эдиториал, 2000.

СОДЕРЖАНИЕ

ПРЕДИСЛОВИЕ	3
I. Структура окна <i>Maple</i>. Арифметические операции, числа, основные константы и стандартные функции. Элементарные преобразования математических выражений	4
§1. Структура окна <i>Maple</i>	4
§2. Арифметические операции, целые и рациональные числа и константы <i>Maple</i>	6
§3. Синтаксис команд. Стандартные функции.....	8
§4. Преобразования математических выражений.....	10
II. Функции в <i>Maple</i>. Операции оценивания. Решение уравнений и неравенств	15
§1. Способы задания функций. Замена переменных.....	15
§2. Операции оценивания.....	19
§3. Решение уравнений.....	21
§4. Решение неравенств.....	24
III. Построение графиков	27
§1. Двумерные графики.....	27
§2. Трехмерные графики. Анимация.....	33
IV. Математический анализ: дифференциальное и интегральное исчисление функции одной переменной	40
§1. Вычисление пределов.....	40
§2. Дифференцирование.....	42
§3. Исследование функции.....	43
§4. Интегрирование.....	53
V. Линейная алгебра	60
§1. Векторная алгебра.....	60
§2. Действия с матрицами.....	62
§3. Спектральный анализ матрицы.....	68
§4. Системы линейных уравнений. Матричные уравнения.....	71
VI. Дифференциальные уравнения	76
§1. Аналитическое решение дифференциальных уравнений.....	76
§2. Численное решение дифференциальных уравнений.....	83
VII. Математический анализ: функции многих переменных, векторный анализ, ряды, интегральные преобразования	92
§1. Дифференциальное исчисление функций многих переменных.....	92
§2. Интегральное исчисление функций многих переменных.....	96
§3. Векторный анализ.....	98

Методы решения математических задач в *Maple*

§4. Ряды и произведения.	102
§5. Интегральные преобразования	108
Рекомендуемая литература.	113
СОДЕРЖАНИЕ.	114